code:
;────────────────────────────────────
;절정계의 처리 등
;────────────────────────────────────
;-------------------------------------------------
;절정 처리
;-------------------------------------------------
@ORGASM_ADD(ARG)
#DIM 합계절정수
#DIM 초조하게해중
#DIM MultipleEc
#DIM Multiplier

VARSET LOCAL
초조하게해중 = 0
LOCAL = ((CUP:ARG:쾌C + PALAM:ARG:쾌C) >= PALAMLV:4) + (CUP:ARG:쾌V + PALAM:ARG:쾌V >= PALAMLV:4) + (CUP:ARG:쾌A + PALAM:ARG:쾌A >= PALAMLV:4) + (CUP:ARG:쾌B + PALAM:ARG:쾌B >= PALAMLV:4) + (CUP:ARG:쾌M + PALAM:ARG:쾌M >= PALAMLV:4)
SELECTCASE LOCAL
	CASE 5
		LOCAL:1 = 5000
	CASE 4
		LOCAL:1 = 3000
	CASE 3
		LOCAL:1 = 2000
	CASE 2
		LOCAL:1 = 1500
	CASE 1
		LOCAL:1 = 1000
ENDSELECT

;시인성의 문제로 남겨 둔다
C = 0
V = 0
A = 0
B = 0
M = 0
;절정 C
;무심코 절정
CALL 절정강도판정(ARG, 0, LOCAL:1)
C = RESULT
;절정 V
CALL 절정강도판정(ARG, 1, LOCAL:1)
V = RESULT

;절정 A
CALL 절정강도판정(ARG, 2, LOCAL:1)
A = RESULT

;절정 B
CALL 절정강도판정(ARG, 3, LOCAL:1)
B = RESULT

;절정 M
CALL 절정강도판정(ARG, 4, LOCAL:1)
M = RESULT
IF TEQUIP:ARG:애태우기모드
	IF 초조하게해발생(ARG, C + V + A + B + M)
		EX:ARG:C절정직전 += C
		EX:ARG:V절정직전 += V
		EX:ARG:A절정직전 += A
		EX:ARG:B절정직전 += B
		EX:ARG:M절정직전 += M
		TCVAR:ARG:초조하게해회수 ++
		초조하게해중 = 1
	ENDIF
ENDIF
SIF 초조하게해중
	GOTO JIRASI
IF CUP:ARG:쾌C
	C += EX:ARG:C절정직전
	EX:ARG:C절정직전 = 0
ENDIF
IF CUP:ARG:쾌V
	V += EX:ARG:V절정직전
	EX:ARG:V절정직전 = 0
ENDIF
IF CUP:ARG:쾌A
	A += EX:ARG:A절정직전
	EX:ARG:A절정직전 = 0
ENDIF
IF CUP:ARG:쾌B
	B += EX:ARG:B절정직전
	EX:ARG:B절정직전 = 0
ENDIF
IF CUP:ARG:쾌M
	M += EX:ARG:M절정직전
	EX:ARG:M절정직전 = 0
ENDIF
;-------------------------------------------------
;절정시의 추가 처리
;-------------------------------------------------
NOWEX:ARG:TotalEX = C + V + A + B + M
MultipleEc = MIN(C, 1) + MIN(V, 1) + MIN(A, 1) + MIN(B, 1) + MIN(M, 1) - 1
POWER Multiplier, 2, MultipleEc
; PRINTFORML {C + V + A + B + M} ec {MultipleEc} {Multiplier}
IF C && V && A && B && M
	C *= 16
	V *= 16
	A *= 16
	B *= 16
	M *= 16
	NOWEX:ARG:오중절정 = 1
ELSEIF C && V && A && B
	C *= 8
	V *= 8
	A *= 8
	B *= 8
	NOWEX:ARG:사중절정 = 1
ELSEIF C && V && A && M
	C *= 8
	V *= 8
	A *= 8
	M *= 8
	NOWEX:ARG:사중절정 = 2
ELSEIF C && V && B && M
	C *= 8
	V *= 8
	B *= 8
	M *= 8
	NOWEX:ARG:사중절정 = 3
ELSEIF C && A && B && M
	C *= 8
	A *= 8
	B *= 8
	M *= 8
	NOWEX:ARG:사중절정 = 4
ELSEIF V && A && B && M
	V *= 8
	A *= 8
	B *= 8
	M *= 8
	NOWEX:ARG:사중절정 = 5
ELSEIF C && V && A
	C *= 4
	V *= 4
	A *= 4
	NOWEX:ARG:삼중절정 = 1
ELSEIF B && V && A
	B *= 4
	V *= 4
	A *= 4
	NOWEX:ARG:삼중절정 = 2
ELSEIF C && B && A
	C *= 4
	B *= 4
	A *= 4
	NOWEX:ARG:삼중절정 = 3
ELSEIF C && V && B
	C *= 4
	V *= 4
	B *= 4
	NOWEX:ARG:삼중절정 = 4
ELSEIF C && V && M
	C *= 4
	V *= 4
	M *= 4
	NOWEX:ARG:삼중절정 = 5
ELSEIF B && V && M
	B *= 4
	V *= 4
	M *= 4
	NOWEX:ARG:삼중절정 = 6
ELSEIF A && V && M
	A *= 4
	V *= 4
	M *= 4
	NOWEX:ARG:삼중절정 = 7
ELSEIF B && C && M
	C *= 4
	B *= 4
	M *= 4
	NOWEX:ARG:삼중절정 = 8
ELSEIF A && B && M
	A *= 4
	B *= 4
	M *= 4
	NOWEX:ARG:삼중절정 = 9
ELSEIF C && A && M
	C *= 4
	A *= 4
	M *= 4
	NOWEX:ARG:삼중절정 = 10
ELSEIF C && V
	C *= 2
	V *= 2
	NOWEX:ARG:이중절정 = 1
ELSEIF C && A
	C *= 2
	A *= 2
	NOWEX:ARG:이중절정 = 2
ELSEIF V && A
	V *= 2
	A *= 2
	NOWEX:ARG:이중절정 = 3
ELSEIF C && B
	C *= 2
	B *= 2
	NOWEX:ARG:이중절정 = 4
ELSEIF V && B
	V *= 2
	B *= 2
	NOWEX:ARG:이중절정 = 5
ELSEIF A && B
	A *= 2
	B *= 2
	NOWEX:ARG:이중절정 = 6
ELSEIF A && M
	A *= 2
	M *= 2
	NOWEX:ARG:이중절정 = 7
ELSEIF V && M
	V *= 2
	M *= 2
	NOWEX:ARG:이중절정 = 8
ELSEIF C && M
	C *= 2
	M *= 2
	NOWEX:ARG:이중절정 = 9
ELSEIF B && M
	B *= 2
	M *= 2
	NOWEX:ARG:이중절정 = 10
ENDIF

IF (C + V + A + B + M) > 0
	TCVAR:ARG:초조하게해회수 = 0
	TEQUIP:ARG:애태우기모드 = 0
ENDIF
합계절정수 = C + V + A + B + M
;시간각인
IF FLAG:70 && ARG
	IF  TALENT:ARG:스침 == 1
		IF 합계절정수 >= 4 && MARK:ARG:시간각인 < 3
			TCVAR:ARG:시간각인취득용 = 3
		ELSEIF 합계절정수 >= 2 && MARK:ARG:시간각인 < 2
			TCVAR:ARG:시간각인취득용 = 2
		ELSEIF  합계절정수 >= 1 && MARK:ARG:시간각인 < 1
			TCVAR:ARG:시간각인취득용 = 1
		ENDIF
	ELSE
		IF 합계절정수 >= 16 && MARK:ARG:시간각인 < 3
			TCVAR:ARG:시간각인취득용 = 3
		ELSEIF 합계절정수 >= 8 && MARK:ARG:시간각인 < 2
			TCVAR:ARG:시간각인취득용 = 2
		ELSEIF  합계절정수 >= 1 && MARK:ARG:시간각인 < 1
			TCVAR:ARG:시간각인취득용 = 1
		ENDIF
	ENDIF
ELSE
;쾌락각인의 처리
;절정의 부위와 강도의 합계가 일정 이상으로 각인을 취득한다
;합계절정수가 8이상으로 쾌락각인 3에 상당
	IF  TALENT:ARG:스침 == 1
		IF 합계절정수 >= 4 && MARK:ARG:쾌락각인 <= 3
			TCVAR:ARG:쾌락강도 = 3
		;합계절정수가 5이상으로 쾌락각인 2에 상당
		ELSEIF 합계절정수 >= 2 && MARK:ARG:쾌락각인 < 2
			TCVAR:ARG:쾌락강도 = 2
		;합계절정수가 2이상으로 쾌락각인 1에 상당
		ELSEIF 합계절정수 >= 1 && MARK:ARG:쾌락각인 < 1
			TCVAR:ARG:쾌락강도 = 1
		ELSE
			TCVAR:ARG:쾌락강도 = 0
		ENDIF
	ELSEIF 합계절정수 >= 24 && MARK:ARG:쾌락각인 <= 3
		TCVAR:ARG:쾌락강도 = 3
	;합계절정수가 12이상으로 쾌락각인 2에 상당
	ELSEIF 합계절정수 >= 12 && MARK:ARG:쾌락각인 < 2
		TCVAR:ARG:쾌락강도 = 2
	;합계절정수가 2이상으로 쾌락각인 1에 상당
	ELSEIF 합계절정수 >= 2 && MARK:ARG:쾌락각인 < 1
		TCVAR:ARG:쾌락강도 = 1
	ELSE
		TCVAR:ARG:쾌락강도 = 0
	ENDIF
ENDIF
IF FLAG:70 == 1
ELSE

IF C
	SOURCE:ARG:노출 += 500 * C
	SOURCE:ARG:굴종 += 200 * C * (SYNCED_ORGASM(ARG) ? 120 # 100) / 100
	SOURCE:ARG:반감 += 40 * C * (CFLAG:ARG:우후후 ?  1 # 3)
	DOWNBASE:ARG:체력 += 20
	DOWNBASE:ARG:기력 += 10
ENDIF
IF V
	SOURCE:ARG:욕정 += 800 * V
	SOURCE:ARG:공순 += 500 * V * (SYNCED_ORGASM(ARG) ? 120 # 100) / 100
	SOURCE:ARG:노출 += 700 * V
	SOURCE:ARG:굴종 += 400 * V * (SYNCED_ORGASM(ARG) ? 120 # 100) / 100
	SOURCE:ARG:반감 += 80 * V * (CFLAG:ARG:우후후 ?  1 # 3)
	DOWNBASE:ARG:체력 += 30
	DOWNBASE:ARG:기력 += 20
ENDIF
IF A
	SOURCE:ARG:욕정 += 500 * A
	SOURCE:ARG:공순 += 700 * A * (SYNCED_ORGASM(ARG) ? 120 # 100) / 100
	SOURCE:ARG:노출 += 900 * A
	SOURCE:ARG:굴종 += 500 * A * (SYNCED_ORGASM(ARG) ? 120 # 100) / 100
	SOURCE:ARG:반감 += 120 * A * (CFLAG:ARG:우후후 ?  1 # 3)
	DOWNBASE:ARG:체력 += 40
	DOWNBASE:ARG:기력 += 20
ENDIF
IF B
	SOURCE:ARG:노출 += 500 * B
	SOURCE:ARG:굴종 += 200 * B * (SYNCED_ORGASM(ARG) ? 120 # 100) / 100
	SOURCE:ARG:반감 += 50 * B * (CFLAG:ARG:우후후 ?  1 # 3)
	DOWNBASE:ARG:체력 += 20
	DOWNBASE:ARG:기력 += 10
ENDIF
IF M
	SOURCE:ARG:사랑 += 1500 * M * (SYNCED_ORGASM(ARG) ? 120 # 100) / 100
	SOURCE:ARG:공순 += 1000 * M * (SYNCED_ORGASM(ARG) ? 120 # 100) / 100
	DOWNBASE:ARG:체력 += 10
	DOWNBASE:ARG:기력 += 5
ENDIF
IF C && V && A && B && M
	DOWNBASE:ARG:체력 -= 80
	DOWNBASE:ARG:기력 -= 70
ELSEIF (C && V && A && B) || (C && V && A && M) || (B && V && A && M) || (C && B && A && M) || (C && V && B && M)
	DOWNBASE:ARG:체력 -= 40
	DOWNBASE:ARG:기력 -= 30

ELSEIF (C && V && A) || (B && V && A) || (C && B && A) || (C && V && B) || (C && V && M) || (B && V && M) || (C && B && M) || (C && A && M) || (V && A && M) || (B && A && M)
	DOWNBASE:ARG:체력 -= 20
	DOWNBASE:ARG:기력 -= 10
ENDIF
ENDIF
;NOWEX에 데이터를 넣는다(절정시 말에 사용한다)
NOWEX:ARG:C절정 = C
NOWEX:ARG:V절정 = V
NOWEX:ARG:A절정 = A
NOWEX:ARG:B절정 = B
NOWEX:ARG:M절정 = M
;절정회수를 늘린다
EX:ARG:C절정 += C
EX:ARG:V절정 += V
EX:ARG:A절정 += A
EX:ARG:B절정 += B
EX:ARG:M절정 += M

;시간정지중 혹은 수면중이라면, 무자각절정회수를 늘린다(다만, 당신을 제외하다)
IF FLAG:70 > 0 || CFLAG:ARG:수면 && ARG
	EX:ARG:무자각C절정 += C
	EX:ARG:무자각V절정 += V
	EX:ARG:무자각A절정 += A
	EX:ARG:무자각B절정 += B
	EX:ARG:무자각M절정 += M
ENDIF

;절정경험을 늘린다
EXP:ARG:절정경험 += 합계절정수

IF TALENT:ARG:스침 == 1 && 합계절정수
	IF ABL:ARG:C감각 < 3 && C
		ABL:ARG:C감각 ++
		CALL COLORMESSAGE(@"%CALLNAME:ARG%의 C감각이 {ABL:ARG:C감각} 가 되었다.",C_YELLOW,2)
	ENDIF
	IF ABL:ARG:B감각 < 3 && B
		ABL:ARG:B감각 ++
		CALL COLORMESSAGE(@"%CALLNAME:ARG%의 B감각이 {ABL:ARG:B감각} 가 되었다.",C_YELLOW,2)
	ENDIF
	IF ABL:ARG:V감각 < 3 && V
		ABL:ARG:V감각 ++
		CALL COLORMESSAGE(@"%CALLNAME:ARG%의 V감각이 {ABL:ARG:V감각} 가 되었다.",C_YELLOW,2)
	ENDIF
	IF ABL:ARG:A감각 < 3 && A
		ABL:ARG:A감각 ++
		CALL COLORMESSAGE(@"%CALLNAME:ARG%의 A감각이 {ABL:ARG:A감각} 가 되었다.",C_YELLOW,2)
	ENDIF
	IF ABL:ARG:M감각 < 3 && M
		ABL:ARG:M감각 ++
		CALL COLORMESSAGE(@"%CALLNAME:ARG%의 M감각이 {ABL:ARG:M감각} 가 되었다.",C_YELLOW,2)
	ENDIF
	IF MAX(GETPALAMLV(PALAM:욕정,5), 합계절정수) > ABL:ARG:욕망 + 1
		ABL:ARG:욕망 ++
		CALL COLORMESSAGE(@"%CALLNAME:ARG%의 욕망이 {ABL:ARG:욕망} 가 되었다.",C_YELLOW,2)
	ENDIF
ENDIF

;체위별 절정경험
IF TEQUIP:ARG:50 > -1
	LOCAL:3 = TEQUIP:ARG:체위
ELSEIF TEQUIP:ARG:51 > -1
	LOCAL:3 = TEQUIP:ARG:체위 + 5
ELSE
	LOCAL:3 = 0
ENDIF
SIF LOCAL:3
	EXP:ARG:(LOCAL:3 + 100) += 1

;절정경험의 추가, 무자각만 당신에게 가산되지 않는다
SIF !SHIRAHU(ARG) && ARG
	EXP:ARG:무자각절정경험 += C + V + A + B + M
EXP:ARG:C절정경험 += C
EXP:ARG:V절정경험 += V
EXP:ARG:A절정경험 += A
EXP:ARG:B절정경험 += B
EXP:ARG:M절정경험 += M

;비디오
IF TEQUIP:ARG:28
	IF NOWEX:ARG:8
		LOCALS = O2/O3/O4/
		TSTR:1 += LOCALS
	ENDIF
	IF NOWEX:ARG:7
		LOCALS = O2/O3/
		TSTR:1 += LOCALS
	ENDIF
	IF NOWEX:ARG:6
		LOCALS = O2/
		TSTR:1 += LOCALS
	ENDIF
	IF NOWEX:ARG:0 == 2
		LOCALS = C/C/
		TSTR:1 += LOCALS
	ELSEIF NOWEX:ARG:0 == 1
		LOCALS = C/
		TSTR:1 += LOCALS
	ENDIF
	IF NOWEX:ARG:1 == 2
		LOCALS = V/V/
		TSTR:1 += LOCALS
	ELSEIF NOWEX:ARG:1 == 1
		LOCALS = V/
		TSTR:1 += LOCALS
	ENDIF
	IF NOWEX:ARG:2 == 2
		LOCALS = A/A/
		TSTR:1 += LOCALS
	ELSEIF NOWEX:ARG:2 == 1
		LOCALS = A/
		TSTR:1 += LOCALS
	ENDIF
	IF NOWEX:ARG:3 == 2
		LOCALS = B/B/
		TSTR:1 += LOCALS
	ELSEIF NOWEX:ARG:3 == 1
		LOCALS = B/
		TSTR:1 += LOCALS
	ENDIF
ENDIF

;絶頂時、利尿剤orおもらし癖で漏らす
IF FLAG:오줌유효 > 0 && MultipleEc > 0 && ( TCVAR:이뇨제 > 0 || TALENT:오줌싸개버릇 > 0 )
	NOWEX:ARG:방뇨 = MultipleEc
	EX:ARG:방뇨 += MultipleEc
	EXP:ARG:방뇨경험 += Multiplier
	;獲得経験値*5 だけ利尿剤の持続時間を短縮
	TCVAR:이뇨제 = MAX( TCVAR:이뇨제 - (Multiplier * 5) , 0)
ENDIF

$JIRASI
;IF 초조하게해중

RETURN 1

@MESSAGE_PALAMCNG_B2
IF NOWEX:오중절정
	IF !GETBIT(CFLAG:3, 2)
		EXP:이상경험 ++
		SETBIT CFLAG:3, 2
	ENDIF
	PRINTL 오중절정
	PRINTL (각각 16배의 구슬을 얻을 수 있습니다)
ELSEIF NOWEX:사중절정
	PRINTL 사중절정
	SELECTCASE NOWEX:사중절정
		CASE 1
			PRINTL C＆V＆A＆B절정
		CASE 2
			PRINTL C＆V＆A＆M절정
		CASE 3
			PRINTL C＆V＆B＆M절정
		CASE 4
			PRINTL C＆A＆B＆M절정
		CASE 5
			PRINTL V＆A＆B＆M절정
	ENDSELECT
	PRINTL (각각 8배의 구슬을 얻을 수 있습니다)
ELSEIF NOWEX:삼중절정
	SELECTCASE NOWEX:삼중절정
		CASE 1
			PRINTL C＆V＆A절정
		CASE 2
			PRINTL B＆V＆A절정
		CASE 3
			PRINTL C＆B＆A절정
		CASE 4
			PRINTL C＆V＆B절정
		CASE 5
			PRINTL C＆V＆M절정
		CASE 6
			PRINTL B＆V＆M절정
		CASE 7
			PRINTL A＆V＆M절정
		CASE 8
			PRINTL C＆B＆M절정
		CASE 9
			PRINTL A＆B＆M절정
		CASE 10
			PRINTL C＆A＆M절정
	ENDSELECT
	PRINTL (각각 4배의 구슬을 얻을 수 있습니다)
ELSEIF NOWEX:이중절정
	SELECTCASE NOWEX:이중절정
		CASE 1
			PRINTL C＆V절정
		CASE 2
			PRINTL C＆A절정
		CASE 3
			PRINTL V＆A절정
		CASE 4
			PRINTL C＆B절정
		CASE 5
			PRINTL V＆B절정
		CASE 6
			PRINTL A＆B절정
		CASE 7
			PRINTL A＆M절정
		CASE 8
			PRINTL V＆M절정
		CASE 9
			PRINTL C＆M절정
		CASE 10
			PRINTL B＆M절정
	ENDSELECT
	PRINTL (각각 2배의 구슬을 얻을 수 있습니다)
ELSEIF NOWEX:B절정
	IF NOWEX:B절정 == 3
		PRINTL 초강 절정 B
	ELSEIF NOWEX:B절정 == 2
		PRINTL 강절정 B
	ELSE
		PRINTL 절정 B
	ENDIF
ELSEIF NOWEX:A절정
	IF NOWEX:A절정 == 3
		PRINTL 초강 절정 A
	ELSEIF NOWEX:A절정 == 2
		PRINTL 강절정 A
	ELSE
		PRINTL 절정 A
	ENDIF
ELSEIF NOWEX:V절정
	IF NOWEX:V절정 == 3
		PRINTL 초강 절정 V
	ELSEIF NOWEX:V절정 == 2
		PRINTL 강절정 V
	ELSE
		PRINTL 절정 V
	ENDIF
ELSEIF NOWEX:C절정
	IF NOWEX:C절정 == 3
		PRINTL 초강 절정 C
	ELSEIF NOWEX:C절정 == 2
		PRINTL 강절정 C
	ELSE
		PRINTL 절정 C
	ENDIF
ELSEIF NOWEX:M절정
	IF NOWEX:M절정 == 3
		PRINTL 초강 절정 M
	ELSEIF NOWEX:M절정 == 2
		PRINTL 강절정 M
	ELSE
		PRINTL 절정 M
	ENDIF
ENDIF

;절정에 의한 욕망 LV업
LOCAL = 0
SIF NOWEX:C절정 >= 2 || NOWEX:B절정 >= 2 || NOWEX:V절정 || NOWEX:A절정
	LOCAL = 1
SIF NOWEX:V절정 >= 2 || NOWEX:A절정 >= 2 || NOWEX:M절정 >= 1
	LOCAL = 3
SIF (NOWEX:C절정 >= 2 || NOWEX:B절정 >= 2) && NOWEX:이중절정
	LOCAL = 3
SIF NOWEX:삼중절정
	LOCAL = 5
SIF NOWEX:사중절정
	LOCAL = 7
SIF NOWEX:오중절정
	LOCAL = 10
;자제심이 있으면(자) 1내린다
SIF TALENT:자제심
	LOCAL -= 1

IF ABL:욕망 < LOCAL
	ABL:욕망 = LOCAL
	PRINTFORML 그리고 욕망이 LV{LOCAL}가 되었다
	;욕망의 상승에 의한[억압]의 소멸을 체크
	CALL YOKUBO_UP_CHECK
ENDIF
PRINTL
;-------------------------------------------------
;조교 대상의 분유체크
;-------------------------------------------------
@TARGET_MILK_CHECK(ARG)
SIF !TALENT:ARG:모유체질
	RETURN 0

LOCAL = CUP:ARG:쾌C / 10 + CUP:ARG:쾌V / 10 + CUP:ARG:쾌A / 10 + CUP:ARG:쾌B * 3
SIF LOCAL == 0
	RETURN 0

;자제심
SIF TALENT:ARG:자제심
	LOCAL /= 2

;쾌감에 솔직
SIF TALENT:ARG:쾌감응답 > 0
	TIMES LOCAL , 1.20

;쾌감에 부정(음란, 창부로 무효)
SIF TALENT:ARG:쾌감응답 < 0 && !TALENT:ARG:음란
	TIMES LOCAL , 0.80

;B민감
SIF TALENT:ARG:B감도 > 0
	TIMES LOCAL , 1.20

;미약
SIF TCVAR:ARG:미약 || TCVAR:ARG:발정
	LOCAL *= 2

;이뇨제
SIF TCVAR:ARG:이뇨제
	LOCAL /= 2

;수면제
SIF TCVAR:ARG:수면제강도
	LOCAL /= 2

;조교자가 유아퇴행
SIF TALENT:PLAYER:애같음
	TIMES LOCAL , 1.20

;조교자가 유치
SIF TALENT:PLAYER:유치
	TIMES LOCAL , 1.20

;작은 가슴
SIF TALENT:ARG:가슴사이즈 < 0
	TIMES LOCAL , 0.50

LOCAL = 100 + LOCAL / 2
SIF TEQUIP:ARG:애태우기모드
	LOCAL = LOCAL * 8 / 10
BASE:ARG:모유 += LOCAL
IF TEQUIP:ARG:애태우기모드 && BASE:ARG:모유 > MAXBASE:ARG:모유 && ARG
	TCVAR:ARG:초조하게해회수 ++
	RETURN 0
ELSEIF  BASE:ARG:모유 > MAXBASE:ARG:모유 * 2
	LOCAL = 1
ELSEIF BASE:ARG:모유 > MAXBASE:ARG:모유
	LOCAL = 0
ELSE
	RETURN 0
ENDIF
LOCAL ++
DOWNBASE:ARG:체력 += 20 * LOCAL
DOWNBASE:ARG:기력 += 40 * LOCAL
IF EXP:ARG:분유경험 < EXPLV:1
	SOURCE:ARG:노출 += 5000 * LOCAL
	SOURCE:ARG:굴종 += 2500 * LOCAL
	SOURCE:ARG:불결 += 1000 * LOCAL
ELSEIF EXP:ARG:분유경험 < EXPLV:2
	SOURCE:ARG:노출 += 2500 * LOCAL
	SOURCE:ARG:굴종 += 1500 * LOCAL
	SOURCE:ARG:불결 += 750 * LOCAL
ELSEIF EXP:ARG:분유경험 < EXPLV:3
	SOURCE:ARG:노출 += 1500 * LOCAL
	SOURCE:ARG:굴종 += 750 * LOCAL
	SOURCE:ARG:불결 += 500 * LOCAL
ELSEIF EXP:ARG:분유경험 < EXPLV:4
	SOURCE:ARG:노출 += 750 * LOCAL
	SOURCE:ARG:굴종 += 400 * LOCAL
	SOURCE:ARG:불결 += 350 * LOCAL
ELSEIF EXP:ARG:분유경험 < EXPLV:5
	SOURCE:ARG:노출 += 500 * LOCAL
	SOURCE:ARG:굴종 += 250 * LOCAL
	SOURCE:ARG:불결 += 200 * LOCAL
ELSE
	SOURCE:ARG:노출 += 250 * LOCAL
	SOURCE:ARG:굴종 += 150 * LOCAL
ENDIF

IF EXP:ARG:분유경험 == 0
	EXP:ARG:이상경험 ++
ENDIF
EXP:ARG:분유경험 += LOCAL
;B에 모유 더러움
STAIN:ARG:5 |= 16
	BASE:ARG:모유 -= MAXBASE:ARG:모유 * LOCAL
SIF BASE:ARG:모유 >= MAXBASE:ARG:모유
	BASE:ARG:모유 = MAXBASE:ARG:모유 - 1
NOWEX:ARG:분유 += LOCAL
EX:ARG:분유 += LOCAL

@TARGET_MILK_MESSAGE(ARG)
CALL COLORMESSAGE(@"《분유》(%CALLNAME:ARG%)", C_PINK, 1,1)
PRINTFORML %CALLNAME:ARG%는 유두로부터\@ NOWEX:ARG:분유 > 1 ?  대량의 # \@모유를 분출했다
IF TEQUIP:ARG:유두빨기
	PRINTFORML %마스터는% 입속에 퍼지는 둥그런 감미를 목을 울리며 다 마셨다
	CALL RECOVER(MASTER, NOWEX:ARG:분유 * 10,"체력")
	CALL RECOVER(MASTER, NOWEX:ARG:분유 * 20,"기력")
	CALL RECOVER(MASTER, NOWEX:ARG:분유 * 10,"정력")
	CALL COLORMESSAGE(@"체력＋{NOWEX:ARG:분유 * 10}", C_AQUA, 1)
	CALL COLORMESSAGE(@"기력＋{NOWEX:ARG:분유 * 20}", C_AQUA, 1)
	CALL COLORMESSAGE(@"정력＋{NOWEX:ARG:분유 * 10}", C_AQUA, 1)
ENDIF

@절정강도판정(ARG, 절정부위, ARG:2)
#DIM 절정부위
#DIM 욕구불만저하계수
#DIM 상승치
; ARG:2=절정 보정
VARSET LOCAL
욕구불만저하계수 = MAX(60 - ABL:ARG:욕망 * 2,30)

SIF FLAG:70
	욕구불만저하계수 /= 2
SIF TCVAR:ARG:발정 || TCVAR:ARG:미약
	욕구불만저하계수 /= 2
LOCAL = (CUP:ARG:절정부위 > 0) * ARG:2
IF CUP:ARG:절정부위 + PALAM:ARG:절정부위 + LOCAL >= PALAMLV:4 * 3
	LOCAL:1 = 3
	CDOWN:ARG:절정부위 = PALAMLV:4 * 3 - LOCAL
ELSEIF CUP:ARG:절정부위 + PALAM:ARG:절정부위 + LOCAL >= PALAMLV:4 * 2
	LOCAL:1 = 2
	CDOWN:ARG:절정부위 = PALAMLV:4 * 2 - LOCAL
ELSEIF CUP:ARG:절정부위 + PALAM:ARG:절정부위 + LOCAL >= PALAMLV:4
	LOCAL:1 = 1
	CDOWN:ARG:절정부위 = PALAMLV:4 - LOCAL
ENDIF
;DOWN:0으로 내려도 절정 이상이라면
;그 값-1이 되도록(듯이) 조정(10000으로 절정이라면 9999)
SIF CUP:ARG:절정부위 + PALAM:ARG:절정부위 - CDOWN:ARG:절정부위 >= PALAMLV:4
	CDOWN:ARG:절정부위 = CUP:ARG:절정부위 + PALAM:ARG:절정부위 - PALAMLV:4 + 1
CFLAG:ARG:욕구불만도 -= LOCAL:1 * 욕구불만저하계수
CFLAG:ARG:욕구불만도 = MAX(CFLAG:ARG:욕구불만도, 0)
IF ARG == MASTER && TARGET >= 1 && (TALENT:ARG:성별 & 1) && !FLAG:70
	CFLAG:ARG:욕구불만도 -= LOCAL:1 * 20
	CFLAG:ARG:욕구불만도 = MAX(CFLAG:ARG:욕구불만도, 0)
ENDIF
상승치 = LOCAL:1
IF !TALENT:ARG:망상의산물
	SIF MAXBASE:ARG:체력 > CSVBASE(ARG, 0) * 2 || MAXBASE:ARG:기력 > CSVBASE(ARG, 1) * 2
		상승치 -= 1
	SIF MAXBASE:ARG:체력 > CSVBASE(ARG, 0) * 3 || MAXBASE:ARG:기력 > CSVBASE(ARG, 1) * 3
		상승치 -= 1
	상승치 = MAX(상승치,0)
	IF ARG
		SIF MAXBASE:ARG:체력 < 5000 || !RAND:2
			MAXBASE:ARG:체력 += 상승치
		SIF MAXBASE:ARG:기력 < 5000 || !RAND:2
			MAXBASE:ARG:기력 += 상승치
	ENDIF
ENDIF
RETURN LOCAL:1

@초조하게해발생(ARG, ARG:1)
#FUNCTION
SIF TCVAR:MASTER:이기초
	RETURNF 0
SIF (ARG:1 + EX:ARG:C절정직전 + EX:ARG:V절정직전 + EX:ARG:A절정직전 + EX:ARG:B절정직전 + EX:ARG:M절정직전) >= (5 - TCVAR:ARG:초조하게해회수) * 2
	RETURNF 0
RETURNF 1

;동시절정 했는지
@SYNCED_ORGASM(ARG)
#FUNCTION
SIF ARG == MASTER
	RETURNF 0
SIF !NOWEX:ARG:C절정 && !NOWEX:ARG:B절정 && !NOWEX:ARG:V절정 && !NOWEX:ARG:A절정 && !NOWEX:ARG:M절정 && !NOWEX:ARG:사정
	RETURNF 0
SIF !NOWEX:MASTER:C절정 && !NOWEX:MASTER:B절정 && !NOWEX:MASTER:V절정 && !NOWEX:MASTER:A절정 && !NOWEX:MASTER:M절정 && !NOWEX:MASTER:사정
	RETURNF 0

RETURNF 1


code:
;FileName_TRACHECK.ERB ----------------------------- Rev1. 00
;V조교의 고통, 반발 보정(피스트 등 격렬한 것은 별처리)
;CALL		USER
;ARG		0:캐릭터 NO
;RETURN		VOID
;COMMENT
;-----------------------------------------------------------
@PAIN_CHECK_V(ARG)
;EXP:V경험을 본다(처녀의 반감은 별도 처리필 모두의로 생략)
IF 삽입불가(ARG) == 1
	TIMES SOURCE:ARG:고통 , 1.50
	TIMES SOURCE:ARG:반감 , 1.20
ELSEIF 삽입불가(ARG) == 2
	TIMES SOURCE:ARG:고통 , 2.50
	TIMES SOURCE:ARG:반감 , 2.00
ENDIF

IF EXP:ARG:V경험 < EXPLV:1
	TIMES SOURCE:ARG:고통 , 2.50
	TIMES SOURCE:ARG:반감 , 1.50
ELSEIF EXP:ARG:V경험 < EXPLV:2
	TIMES SOURCE:ARG:고통 , 1.50
	TIMES SOURCE:ARG:반감 , 1.20
ELSEIF EXP:ARG:V경험 < EXPLV:3
	TIMES SOURCE:ARG:고통 , 1.00
ELSEIF EXP:ARG:V경험 < EXPLV:4
	TIMES SOURCE:ARG:고통 , 0.50
ELSEIF EXP:ARG:V경험 < EXPLV:5
	TIMES SOURCE:ARG:고통 , 0.30
ELSEIF EXP:ARG:V경험 >= EXPLV:5
	TIMES SOURCE:ARG:고통 , 0.10
ENDIF

;PALAM:윤활을 본다
IF PALAM:ARG:윤활 < PALAMLV:1
	TIMES SOURCE:ARG:고통 , 1.20
	TIMES SOURCE:ARG:반감 , 1.50
ELSEIF PALAM:ARG:윤활 < PALAMLV:2
	TIMES SOURCE:ARG:고통 , 0.70
	TIMES SOURCE:ARG:반감 , 1.20
ELSEIF PALAM:ARG:윤활 < PALAMLV:3
	TIMES SOURCE:ARG:고통 , 0.40
ELSEIF PALAM:ARG:윤활 < PALAMLV:4
	TIMES SOURCE:ARG:고통 , 0.10
ELSEIF PALAM:ARG:윤활 >= PALAMLV:4
	TIMES SOURCE:ARG:고통 , 0.10
ENDIF

;FileName_TRACHECK.ERB ----------------------------- Rev1. 00
;애널 조교의 고통, 반발 보정
;CALL		USER
;ARG		0:캐릭터 NO
;RETURN		VOID
;COMMENT
;-----------------------------------------------------------
@PAIN_CHECK_A(ARG)
IF 삽입불가(ARG) == 1
	TIMES SOURCE:ARG:고통 , 1.50
	TIMES SOURCE:ARG:반감 , 1.20
ELSEIF 삽입불가(ARG) == 2
	TIMES SOURCE:ARG:고통 , 2.50
	TIMES SOURCE:ARG:반감 , 2.00
ENDIF
;EXP:A경험을 본다
IF EXP:ARG:A경험 < EXPLV:1
	TIMES SOURCE:ARG:고통 , 3.00
	TIMES SOURCE:ARG:반감 , 3.00
ELSEIF EXP:ARG:A경험 < EXPLV:2
	TIMES SOURCE:ARG:고통 , 2.00
	TIMES SOURCE:ARG:반감 , 2.00
ELSEIF EXP:ARG:A경험 < EXPLV:3
	TIMES SOURCE:ARG:고통 , 1.00
	TIMES SOURCE:ARG:반감 , 1.50
ELSEIF EXP:ARG:A경험 < EXPLV:4
	TIMES SOURCE:ARG:고통 , 0.60
	TIMES SOURCE:ARG:반감 , 1.20
ELSEIF EXP:ARG:A경험 < EXPLV:5
	TIMES SOURCE:ARG:고통 , 0.40
ELSEIF EXP:ARG:A경험 >= EXPLV:5
	TIMES SOURCE:ARG:고통 , 0.20
ENDIF

;PALAM:윤활을 본다
IF PALAM:ARG:윤활 < PALAMLV:1
	TIMES SOURCE:ARG:고통 , 1.20
	TIMES SOURCE:ARG:반감 , 2.00
ELSEIF PALAM:ARG:윤활 < PALAMLV:2
	TIMES SOURCE:ARG:고통 , 1.00
	TIMES SOURCE:ARG:반감 , 1.50
ELSEIF PALAM:ARG:윤활 < PALAMLV:3
	TIMES SOURCE:ARG:고통 , 0.60
ELSEIF PALAM:ARG:윤활 < PALAMLV:4
	TIMES SOURCE:ARG:고통 , 0.30
ELSEIF PALAM:ARG:윤활 >= PALAMLV:4
	TIMES SOURCE:ARG:고통 , 0.20
ENDIF

repo_name: DNS/eraTWKR

license: mit


code:
;FileName_TRACHECK.ERB ----------------------------- Rev1. 00
;시간경과
;CALL		USER
;ARG		VOID
;RETURN		VOID
;COMMENT	SOURCE_CHECK가 달릴 때마다 5분 계상
;-----------------------------------------------------------
@TRACHECK_TIME
;귀찮아서의로 5 분 간격
IF FLAG:70 == 1
ELSE
SIF SELECTCOM < 200
	TIME += 5
ENDIF
;FileName_TRACHECK.ERB ----------------------------- Rev1. 00
;호감도 체크
;CALL		USER
;ARG		0:캐릭터 NO, 1:소스 NO
;RETURN		호감도보너스 계수
;COMMENT
;-----------------------------------------------------------
@MASTER_FAVOR_CHECK(ARG, ARG:1)
#FUNCTION
SIF ASSIPLAY
	RETURNF 100
SELECTCASE ARG:1
	CASE 0 TO 3
		RETURNF GET_REVISION(CFLAG:ARG:2, 200 , 20000) + 100
	CASE 10
		RETURNF GET_REVISION(CFLAG:ARG:2, 200 , 15000) + 100
	CASE 30 TO 33
		RETURNF MAX(50,100 - CFLAG:ARG:2 / 200)
	CASEELSE
		RETURNF 100
ENDSELECT

;FileName_TRACHECK.ERB ----------------------------- Rev1. 00
;기교의 체크
;CALL		USER
;ARG		0:캐릭터 NO, 1:소스 NO
;RETURN		기교 보너스 계수
;COMMENT
;-----------------------------------------------------------
@TECHNIQUE_CHECK(ARG, ARG:1)
#FUNCTION
SIF TCVAR:ARG:116 < 0
	RETURNF 100
SELECTCASE ARG:1
	CASE 0 TO 3
		RETURNF GET_REVISION(ABL:(TCVAR:ARG:116):기교 + 2, 500, 10)
	CASEELSE
		RETURNF 100
ENDSELECT

;FileName_TRACHECK.ERB ----------------------------- Rev1. 00
;무드 체크
;CALL		USER
;ARG		0:캐릭터 NO, 1:소스 NO
;RETURN		무드 보너스치
;COMMENT
;-----------------------------------------------------------
@MOOD_CHECK(ARG, ARG:1)
#FUNCTION
SELECTCASE ARG:1
	CASE 10,12,15,16,18,20,21,22
		RETURNF GET_REVISION(BASE:ARG:무드, 50 , 500) + 100
	CASE 30 TO 33
		RETURNF 200 - BASE:ARG:무드 / 100
	CASEELSE
		RETURNF 100
ENDSELECT

;FileName_TRACHECK.ERB ----------------------------- Rev1. 00
;이성 체크
;CALL		USER
;ARG		0:캐릭터 NO, 1:소스 NO
;RETURN		이성 보너스치
;COMMENT
;-----------------------------------------------------------
@REASON_CHECK(ARG, ARG:1)
#FUNCTION
SELECTCASE ARG:1
	CASE 0,1,2,3,10,12,15,16,18
		RETURNF GET_REVISION(1000 - BASE:ARG:이성, 50 , 500) + 100
	CASE 30 TO 33
		RETURNF 50 + BASE:ARG:이성 / 100
	CASEELSE
		RETURNF 100
ENDSELECT
;FileName_TRACHECK.ERB ----------------------------- Rev1. 00
;동성의 경우의 체크
;CALL		USER
;ARG		0:캐릭터 NO
;RETURN		VOID
;COMMENT
;-----------------------------------------------------------
@SOURCE_SEX_CHECK(ARG)
#DIM 레즈호모끼
#DIM 레즈호모중독
#DIM 당신레즈호모끼
#DIM 당신레즈호모중독
SIF !HETEROSEX(ARG, PLAYER) || TALENT:ARG:81 == -1
	RETURN 0
;동성애
레즈호모끼 = ABL:ARG:(16 + HETEROSEX(ARG, PLAYER))
;ABL:레즈호모끼를 본다
SELECTCASE 레즈호모끼
	CASE 0 TO 5
		SOURCE:ARG:달성 = SOURCE:ARG:달성 * (레즈호모끼 + 10) / 20
		SOURCE:ARG:유혹 = SOURCE:ARG:유혹 * (레즈호모끼 + 10) / 25
		SOURCE:ARG:송구스러운 = SOURCE:ARG:송구스러운 * (레즈호모끼 + 10) / 20
		SOURCE:ARG:도발 = SOURCE:ARG:도발 * (레즈호모끼 + 10) / 25
		SOURCE:ARG:굴종 = SOURCE:ARG:굴종 * (레즈호모끼 + 10) / 20
		SOURCE:ARG:사랑 = SOURCE:ARG:사랑 * (레즈호모끼 + 10) / 20
		SOURCE:ARG:일탈 = SOURCE:ARG:일탈 * (30 - 2 * 레즈호모끼) / 10
	CASE 6 TO 10
		SOURCE:ARG:달성 = SOURCE:ARG:달성 * (레즈호모끼 * 2 + 20) / 40
		SOURCE:ARG:유혹 = SOURCE:ARG:유혹 * (레즈호모끼 * 2 + 20) / 50
		SOURCE:ARG:송구스러운 = SOURCE:ARG:송구스러운 * (레즈호모끼 * 2 + 20) / 40
		SOURCE:ARG:도발 = SOURCE:ARG:도발 * (레즈호모끼 * 2 + 20) / 50
		SOURCE:ARG:굴종 = SOURCE:ARG:굴종 * (레즈호모끼 * 2 + 20) / 40
		SOURCE:ARG:사랑 = SOURCE:ARG:사랑 * (레즈호모끼 * 2 + 20) / 40
		SOURCE:ARG:일탈 = SOURCE:ARG:일탈 * (55 - 3 * 레즈호모끼) / 20
	CASEELSE
		TIMES SOURCE:ARG:달성 , 1.10
		TIMES SOURCE:ARG:유혹 , 1.10
		TIMES SOURCE:ARG:송구스러운 , 1.20
		TIMES SOURCE:ARG:도발 , 1.10
		TIMES SOURCE:ARG:굴종 , 1.20
		TIMES SOURCE:ARG:일탈 , 1.00
ENDSELECT

;ABL:중독을 본다
레즈호모중독 = ABL:ARG:(31 + HETEROSEX(ARG, PLAYER))
SELECTCASE 레즈호모중독
	CASE 0 TO 3
		SOURCE:ARG:쾌C = SOURCE:ARG:쾌C * (레즈호모중독 + 5) / 10
		SOURCE:ARG:쾌V = SOURCE:ARG:쾌V * (레즈호모중독 + 5) / 10
		SOURCE:ARG:쾌A = SOURCE:ARG:쾌A * (레즈호모중독 + 5) / 10
		SOURCE:ARG:쾌B = SOURCE:ARG:쾌B * (레즈호모중독 + 5) / 10
		SOURCE:ARG:쾌M = SOURCE:ARG:쾌M * (레즈호모중독 + 5) / 10
		SOURCE:ARG:여쾌C = SOURCE:ARG:여쾌C * (레즈호모중독 + 5) / 10
		SOURCE:ARG:여쾌V = SOURCE:ARG:여쾌V * (레즈호모중독 + 5) / 10
		SOURCE:ARG:여쾌A = SOURCE:ARG:여쾌A * (레즈호모중독 + 5) / 10
		SOURCE:ARG:여쾌B = SOURCE:ARG:여쾌B * (레즈호모중독 + 5) / 10
		SOURCE:ARG:여쾌M = SOURCE:ARG:여쾌M * (레즈호모중독 + 5) / 10
		SOURCE:ARG:불결 = SOURCE:ARG:불결 * (20 - 2 * 레즈호모중독) / 10
	CASE 4 TO 7
		SOURCE:ARG:쾌C = SOURCE:ARG:쾌C * (레즈호모중독 + 12) / 20
		SOURCE:ARG:쾌V = SOURCE:ARG:쾌V * (레즈호모중독 + 12) / 20
		SOURCE:ARG:쾌A = SOURCE:ARG:쾌A * (레즈호모중독 + 12) / 20
		SOURCE:ARG:쾌B = SOURCE:ARG:쾌B * (레즈호모중독 + 12) / 20
		SOURCE:ARG:쾌M = SOURCE:ARG:쾌M * (레즈호모중독 + 12) / 20
		SOURCE:ARG:여쾌C = SOURCE:ARG:여쾌C * (레즈호모중독 + 12) / 20
		SOURCE:ARG:여쾌V = SOURCE:ARG:여쾌V * (레즈호모중독 + 12) / 20
		SOURCE:ARG:여쾌A = SOURCE:ARG:여쾌A * (레즈호모중독 + 12) / 20
		SOURCE:ARG:여쾌B = SOURCE:ARG:여쾌B * (레즈호모중독 + 12) / 20
		SOURCE:ARG:여쾌M = SOURCE:ARG:여쾌M * (레즈호모중독 + 12) / 20
		SOURCE:ARG:불결 = SOURCE:ARG:불결 * (30 - 레즈호모중독) / 20
	CASEELSE
ENDSELECT
	;조교자의 ABL:레즈호모끼
	당신레즈호모끼 = ABL:PLAYER:(16 + HETEROSEX(ARG, PLAYER))
SELECTCASE 당신레즈호모끼
	CASE 0 TO 5
		SOURCE:ARG:사랑 = SOURCE:ARG:사랑 * (당신레즈호모끼 + 15) / 20
		SOURCE:ARG:달성 = SOURCE:ARG:달성 * (당신레즈호모끼 + 15) / 20
	CASE 6 TO 10
		SOURCE:ARG:사랑 = SOURCE:ARG:사랑 * (당신레즈호모끼 + 35) / 40
		SOURCE:ARG:달성 = SOURCE:ARG:달성 * (당신레즈호모끼 + 35) / 40
	CASEELSE
		TIMES SOURCE:ARG:사랑 , 1.20
		TIMES SOURCE:ARG:달성 , 1.20
ENDSELECT
;조교자의 ABL:중독
당신레즈호모중독 = ABL:PLAYER:(31 + HETEROSEX(ARG, PLAYER))
SELECTCASE 당신레즈호모중독
	CASE 0 TO 7
		SOURCE:ARG:사랑 = SOURCE:ARG:사랑 * (당신레즈호모끼 + 20) / 20
		SOURCE:ARG:달성 = SOURCE:ARG:달성 * (당신레즈호모끼 + 20) / 20
		SOURCE:ARG:성행동 = SOURCE:ARG:성행동 * (당신레즈호모끼 + 20) / 20
	CASEELSE
		TIMES SOURCE:ARG:사랑 , 1.40
		TIMES SOURCE:ARG:성행동 , 1.40
		TIMES SOURCE:ARG:달성 , 1.40
ENDSELECT
;조교자의 자제심
IF TALENT:PLAYER:자제심
	TIMES SOURCE:ARG:성행동 , 0.50
	TIMES SOURCE:ARG:달성 , 0.50
ENDIF

;FileName_TRACHECK.ERB ----------------------------- Rev1. 00
;PLAYER 능력의 처리
;CALL		USER
;ARG		0:캐릭터 NO
;RETURN		VOID
;COMMENT
;-----------------------------------------------------------
@PLAYER_SKILL_CHECK(ARG)
;조교자의 TALENT:해방을 본다
IF TALENT:PLAYER:자기애 > 0
	TIMES SOURCE:ARG:쾌C , 1.20
	TIMES SOURCE:ARG:쾌V , 1.20
	TIMES SOURCE:ARG:쾌A , 1.20
	TIMES SOURCE:ARG:쾌B , 1.20
	TIMES SOURCE:ARG:사랑 , 1.20
ENDIF
;조교자의 TALENT:양도를 본다
IF TALENT:PLAYER:성별기호 == -1 && HETEROSEX(PLAYER, ARG)
	TIMES SOURCE:ARG:사랑 , 1.20
	TIMES SOURCE:ARG:성행동 , 1.20
	TIMES SOURCE:ARG:달성 , 1.20
ENDIF
;조교자의 TALENT:새드를 본다
SIF TALENT:PLAYER:새드
	TIMES SOURCE:ARG:공포 , 1.50
;조교자의 TALENT:소악마를 본다
SIF TALENT:PLAYER:소악마
	TIMES SOURCE:ARG:노출 , 1.50
;조교자의 TALENT:매혹을 본다
IF TALENT:PLAYER:매혹
	TIMES SOURCE:ARG:불결 , 0.50
	TIMES SOURCE:ARG:일탈 , 0.50

	TIMES SOURCE:ARG:쾌C , 1.20
	TIMES SOURCE:ARG:쾌V , 1.20
	TIMES SOURCE:ARG:쾌A , 1.20
	TIMES SOURCE:ARG:쾌B , 1.20
	TIMES SOURCE:ARG:쾌M , 1.20
	TIMES SOURCE:ARG:성행동 , 1.20
ENDIF
;조교자의 TALENT:수수께끼의매력을 본다
IF TALENT:PLAYER:수수께끼의매력
	TIMES SOURCE:ARG:불결 , 0.50
	TIMES SOURCE:ARG:일탈 , 0.50
	TIMES SOURCE:ARG:쾌C , 1.20
	TIMES SOURCE:ARG:쾌V , 1.20
	TIMES SOURCE:ARG:쾌A , 1.20
	TIMES SOURCE:ARG:쾌B , 1.20
	TIMES SOURCE:ARG:쾌M , 1.20
	TIMES SOURCE:ARG:성행동 , 1.20
	TIMES SOURCE:ARG:달성 , 1.50
ENDIF

;FileName_TRACHECK.ERB ----------------------------- Rev1. 00
;순종의 상승에 의한[저항]의 소멸을 체크
;CALL		USER
;ARG		VOID
;RETURN		VOID
;COMMENT
;-----------------------------------------------------------
@JUJUN_UP_CHECK
IF FLAG:70 == 1 || CFLAG:장난
ELSE
FOR LOCAL, 0, CHARANUM
	IF ABL:LOCAL:순종 >= 7 && TALENT:LOCAL:저항
		PRINTFORM %NAME:LOCAL%는 [저항]을 잃었다
		TALENT:LOCAL:저항 = 0
		PRINTL 부정의 구슬이 절반으로 줄었다
		JUEL:LOCAL:100 /= 2
	ENDIF
NEXT
ENDIF
;FileName_TRACHECK.ERB ----------------------------- Rev1. 00
;욕망의 상승에 의한[억압]의 소멸을 체크
;CALL		USER
;ARG		VOID
;RETURN		VOID
;COMMENT
;-----------------------------------------------------------
@YOKUBO_UP_CHECK
IF FLAG:70 == 1 || CFLAG:장난
ELSE
FOR LOCAL, 0, CHARANUM
	IF ABL:LOCAL:욕망 >= 7 && TALENT:LOCAL:자기애 < 0
		PRINTFORM %NAME:LOCAL%는 [억압]을 잃었다
		TALENT:LOCAL:자기애 = 0
		PRINTL 부정의 구슬이 절반을 줄었다
		JUEL:LOCAL:100 /= 2
	ENDIF
NEXT
ENDIF
;FileName_TRACHECK.ERB ----------------------------- Rev1. 00
;고통쾌락경험, 봉사쾌락경험의 체크
;CALL		USER
;ARG		0:캐릭터 NO
;RETURN		VOID
;COMMENT
;-----------------------------------------------------------
@EXP_GOT_CHECK(ARG)
#DIM 기학증가량
IF FLAG:70 == 1 || CFLAG:장난
ELSE
;봉사계인, 봉사 쾌락의 체크
LOCAL = CUP:ARG:쾌C + CUP:ARG:쾌V + CUP:ARG:쾌A + CUP:ARG:쾌B + CUP:ARG:쾌M + CUP:ARG:공순 / 5
LOCAL:1 = 0

SELECTCASE CUP:ARG:욕정
	CASE IS < 100
		LOCAL = 0
	CASE 101 TO 1000
	CASE 1001 TO 5000
		LOCAL *= 2
	CASE 5001 TO 10000
		LOCAL *= 3
	CASE 10001 TO 30000
		LOCAL *= 4
	CASEELSE
		LOCAL *= 5
ENDSELECT

LOCAL += TCVAR:ARG:사정게이지증가량
SIF TCVAR:ARG:기학플래그
	LOCAL /= 4
;봉사플래그가 서 있다
IF TCVAR:ARG:108
	;봉사 쾌락의 체크
	IF LOCAL >= 100000
		LOCAL:1 = 16
		TIMES CUP:ARG:공포 , 0.65
		TIMES CUP:ARG:반감 , 0.30
	ELSEIF LOCAL >= 50000
		LOCAL:1 = 12
		TIMES CUP:ARG:공포 , 0.70
		TIMES CUP:ARG:반감 , 0.40
	ELSEIF LOCAL >= 30000
		LOCAL:1 = 8
		TIMES CUP:ARG:공포 , 0.75
		TIMES CUP:ARG:반감 , 0.50
	ELSEIF LOCAL >= 10000
		LOCAL:1 = 4
		TIMES CUP:ARG:공포 , 0.80
		TIMES CUP:ARG:반감 , 0.60
	ELSEIF LOCAL >= 5000
		LOCAL:1 = 2
		TIMES CUP:ARG:공포 , 0.85
		TIMES CUP:ARG:반감 , 0.70
	ELSEIF LOCAL >= 3000
		LOCAL:1 = 1
		TIMES CUP:ARG:공포 , 0.90
		TIMES CUP:ARG:반감 , 0.80
	ENDIF
	EXP:ARG:봉사쾌락경험 += LOCAL:1
ENDIF

;상대의 기학 쾌락의 체크
IF TCVAR:ARG:기학플래그
	LOCAL = CUP:ARG:쾌C + CUP:ARG:쾌V + CUP:ARG:쾌A + CUP:ARG:쾌B + CUP:ARG:쾌M
	SELECTCASE CUP:ARG:우월
		CASE IS < 100
			LOCAL = 0
		CASE 101 TO 1000
		CASE 1001 TO 5000
			LOCAL *= 2
		CASE 5001 TO 10000
			LOCAL *= 3
		CASE 10001 TO 30000
			LOCAL *= 4
		CASEELSE
			LOCAL *= 5
	ENDSELECT
	;가학 소스가 없으면 반감
	SIF !SOURCE:ARG:가학
		LOCAL /= 2
	;봉사쾌락경험이 들어가는 경우 감소
	SIF TCVAR:ARG:108
		LOCAL /= 4
	SELECTCASE LOCAL
		CASE IS > 100000
			기학증가량 = 16
		CASE IS > 50000
			기학증가량 = 12
		CASE IS > 30000
			기학증가량 = 8
		CASE IS > 10000
			기학증가량 = 4
		CASE IS > 5000
			기학증가량 = 2
		CASE IS > 3000
			기학증가량 = 1
	ENDSELECT
	SIF ABL:ARG:새드끼 < ABL:MASTER:새드끼
		기학증가량 /= 2
	EXP:ARG:가학쾌락경험 += 기학증가량
	TCVAR:ARG:기학플래그 = 0
ENDIF
;고통 쾌락의 체크
LOCAL = 0
IF CUP:ARG:욕정 >= 3000 && CUP:ARG:고통 >= 2000
	LOCAL = 16
	TIMES CUP:ARG:공포 , 0.65
ELSEIF CUP:ARG:욕정 >= 2500 && CUP:ARG:고통 >= 1500
	LOCAL = 12
	TIMES CUP:ARG:공포 , 0.70
ELSEIF CUP:ARG:욕정 >= 1500 && CUP:ARG:고통 >= 1000
	LOCAL = 8
	TIMES CUP:ARG:공포 , 0.75
ELSEIF CUP:ARG:욕정 >= 1000 && CUP:ARG:고통 >= 500
	LOCAL = 4
	TIMES CUP:ARG:공포 , 0.80
ELSEIF CUP:ARG:욕정 >= 600 && CUP:ARG:고통 >= 300
	LOCAL = 2
	TIMES CUP:ARG:공포 , 0.85
ELSEIF CUP:ARG:욕정 >= 300 && CUP:ARG:고통 >= 100
	LOCAL = 1
	TIMES CUP:ARG:공포 , 0.90
ENDIF
IF LOCAL > 0 && ARG != PLAYER
	EXP:ARG:고통쾌락경험 += LOCAL

	;가학쾌락경험
	LOCAL = ABL:PLAYER:새드끼 + ABL:ARG:마조끼 - 1
	SIF LOCAL > 0
		EXP:PLAYER:가학쾌락경험 += LOCAL
ENDIF
ENDIF
;FileName_TRACHECK.ERB ----------------------------- Rev1. 00
;나무삼 카드 발행 체크
;CALL		USER
;ARG		0:캐릭터 NO
;RETURN		VOID
;COMMENT
;-----------------------------------------------------------
@CARD_GOT_CHECK(ARG)
;분노 상태로 우후후 상태가 아니고 오늘중에 경고장을 받지 않았다
IF CFLAG:ARG:빡침 && !CFLAG:ARG:우후후 && !CFLAG:ARG:328
	IF ARG != NO:1
		PRINTFORML %CALLNAME:ARG%는 레이무에게 %CALLNAME:MASTER%의 행위를 보고했다…
	ELSE
		PRINTFORML %CALLNAME:ARG%는 %CALLNAME:MASTER%에 부적을 들이밀었다
	ENDIF
	IF SELECTCOM == 333
		PRINTFORML 자신은 %CALLNAME:ARG%의 푸념을 들었을 뿐이라고 변명하자 레이무는 넘어가주었다
		GOTO SKIP
	ENDIF
	PRINTFORML %마스터는% 경고장을 취득했습니다
	;카드 매수 추가
	CFLAG:MASTER:329 ++
	;1날 1회 밖에 카드는 발행할 수 없다
	CFLAG:ARG:328 = 1
	PRINTL
	PRINTFORML 현재의 경고장  {CFLAG:MASTER:329}장
	IF CFLAG:MASTER:329 > 3
		PRINTL
		PRINTW …
		PRINTW ……
		PRINTFORML 「네가 있어서 재미있었어. 하지만 도가 지나친 것 같아」
		PRINTFORMW ……
		PRINTL
		PRINTL
		PRINTFORML %마스터는% 기억도 능력도 잃고 환상향에서 쫓겨납니다 (게임 오버?)
		PRINTL [0] 용서해주세요! (Continue)  [1] 마음을 고쳐먹겠습니다! (Exit)
		$GAME_OVER
		INPUT
		IF RESULT == 0
			PRINTL 「……정말로, 반성 중이야?」
			PRINTFORML 연방 끄덕이는 %CALLNAME:MASTER%의 모습에 레이무는 한숨을 내쉰다
			PRINTW 「다음에도 용서받을 거라고는 생각하지 마」
			PRINTL
			PRINTFORML 경고장이 1장 무효가 되었습니다.
			CFLAG:MASTER:329 = CFLAG:MASTER:329 - 1
			PRINTFORMW 현재의 경고장  {CFLAG:MASTER:329}장
		ELSEIF RESULT == 1
			PRINTFORML 정신이 드니 %마스터는% 마을 한구석에 서 있었다
			PRINTFORML 여러 가지로 즐거운 일이 있었던 느낌이지만 기분 탓일 것이다
			PRINTFORML %마스터는% 일상으로 돌아갔다
			PRINTW 그 뒤 치한으로 체포되었다, 같은 일은 있을지도 모르고 없을지도 모른다 (게임 오버)
			QUIT
		ELSE
			CLEARLINE 1
			GOTO GAME_OVER
		ENDIF
	ENDIF
	$SKIP
ENDIF


code:
;FileName_TRACHECK.ERB ----------------------------- Rev1. 00
;각인 취득의 체크
;CALL		USER
;ARG		0:캐릭터 NO
;RETURN		VOID
;COMMENT
;COMMENT	21/08/27 반발각인 취득 사양을 일부 변경
;-----------------------------------------------------------
@MARK_GOT_CHECK(ARG)
#DIM 각인취득치
;-------------------------------------------------
;고통각인
;-------------------------------------------------
IF CUP:ARG:고통 >= 2000 && CUP:ARG:고통 < 4000 && MARK:ARG:고통각인 <= 0
	MARK:ARG:고통각인 = 1
	TFLAG:고통각인취득 = 1
ELSEIF CUP:ARG:고통 >= 4000 && CUP:ARG:고통 < 8000 && MARK:ARG:고통각인 <= 1
	MARK:ARG:고통각인 = 2
	TFLAG:고통각인취득 = 2
	;다부짐·감정결여의 경우는 고통각인에 의한 순종 업은 없음
	IF !ABL:ARG:순종 && TALENT:ARG:담력 > 0 && !TALENT:ARG:감정결여
		ABL:ARG:순종 = 1
		TFLAG:각인순종변화 = 1
	ENDIF
ELSEIF CUP:ARG:고통 >= 8000 && MARK:ARG:고통각인 <= 2
	MARK:ARG:고통각인 = 3
	TFLAG:고통각인취득 = 3
	;다부짐·감정결여의 경우는 고통각인에 의한 순종 업은 없음
	IF !ABL:ARG:순종 && TALENT:ARG:담력 > 0 && !TALENT:ARG:감정결여
		ABL:ARG:순종 = 1
		TFLAG:각인순종변화 = 1
	ENDIF
	IF TALENT:PLAYER:새드
		EXP:ARG:이상경험 += 2
	ELSEIF TALENT:PLAYER:바늘질 && SELECTCOM == 103
		EXP:ARG:이상경험 += 2
	ELSE
		EXP:ARG:이상경험 += 1
	ENDIF
ENDIF
;-------------------------------------------------
;시간각인
;-------------------------------------------------
IF TCVAR:ARG:시간각인취득용 == 1 && MARK:ARG:시간각인 <= 0
	MARK:ARG:시간각인 = 1
	TFLAG:시간각인취득 = 1
ELSEIF TCVAR:ARG:시간각인취득용 == 2 && MARK:ARG:시간각인 <= 1
	MARK:ARG:시간각인 = 2
	TFLAG:시간각인취득 = 2
ELSEIF TCVAR:ARG:시간각인취득용 == 3 && MARK:ARG:시간각인 <= 2
	MARK:ARG:시간각인 = 3
	TFLAG:시간각인취득 = 3
ENDIF
;-------------------------------------------------
;쾌락각인
;-------------------------------------------------
IF TCVAR:ARG:쾌락강도 == 1 && MARK:ARG:쾌락각인 <= 0
	MARK:ARG:쾌락각인 = 1
	TFLAG:23 = 1
ELSEIF TCVAR:쾌락강도 == 2 && MARK:ARG:쾌락각인 <= 1
	MARK:ARG:쾌락각인 = 2
	TFLAG:23 = 2
ELSEIF TCVAR:쾌락강도 == 3 && MARK:ARG:쾌락각인 <= 2
	MARK:ARG:쾌락각인 = 3
	TFLAG:23 = 3
	;자제심·감정결여의 경우는 쾌락각인에 의한 순종 업은 없음
	IF !ABL:ARG:순종 && !TALENT:ARG:자제심 && !TALENT:ARG:감정결여
		ABL:ARG:순종 = 1
		TFLAG:각인순종변화 = 1
	ENDIF
;ELSEIF  TCVAR:쾌락강도 == 3 && TALENT:유치 && !TALENT:초경
;		TALENT:초경 = 1
;		TFLAG:23 = 4
ENDIF
;-------------------------------------------------
;괘씸각인
;-------------------------------------------------
IF FLAG:70 == 1 || CFLAG:ARG:장난
ELSE
[SKIPSTART]
	A = CUP:ARG:쾌C + CUP:ARG:쾌V + CUP:ARG:쾌B + CUP:ARG:쾌A

	;쾌락에 의한 LvUP는 10번+취득한 괘씸각인의 레벨의 TFLAG를 준다
	IF A >= 3000 && A < 5000 && MARK:ARG:쾌락각인 > MARK:ARG:괘씸각인 && MARK:ARG:괘씸각인 <= 0
		MARK:ARG:괘씸각인 = 1
		TFLAG:24 = 11
	ELSEIF A >= 5000 && A < 9000 && MARK:ARG:쾌락각인 > MARK:ARG:괘씸각인 && MARK:ARG:괘씸각인 <= 1
		MARK:ARG:괘씸각인 = 2
		TFLAG:24 = 12
		;감정결여의 경우는 괘씸각인에 의한 순종 업은 없음
		IF !ABL:ARG:순종 && !TALENT:ARG:감정결여
			ABL:ARG:순종 = 1
			TFLAG:각인순종변화 = 1
		ENDIF
	ELSEIF A >= 9000 && MARK:ARG:쾌락각인 > MARK:ARG:괘씸각인 && MARK:ARG:괘씸각인 <= 2
		MARK:ARG:괘씸각인 = 3
		TFLAG:24 = 13
		;감정결여의 경우는 괘씸각인에 의한 순종 업은 없음
		IF ABL:ARG:순종 <= 1 && !TALENT:ARG:감정결여
			ABL:ARG:순종 = 2
			TFLAG:각인순종변화 = 1
		ENDIF
	ENDIF
	A = CUP:ARG:고통 + CUP:ARG:공포

	;고통과 공포에 의한 LvUP는 20번+취득한 괘씸각인의 레벨의 TFLAG를 준다
	IF A >= 1500 && A < 2400 && MARK:ARG:괘씸각인 <= 0
		MARK:ARG:괘씸각인 = 1
		TFLAG:24 = 21
	ELSEIF A >= 2400 && A < 6000 && MARK:ARG:괘씸각인 <= 1
		MARK:ARG:괘씸각인 = 2
		TFLAG:24 = 22
		;감정결여의 경우는 괘씸각인에 의한 순종 업은 없음
		IF !ABL:ARG:순종 && !TALENT:ARG:감정결여
			ABL:ARG:순종 = 1
			TFLAG:각인순종변화 = 1
		ENDIF
	ELSEIF A >= 6000 && MARK:ARG:괘씸각인 <= 2
		MARK:ARG:괘씸각인 = 3
		TFLAG:24 = 23
		;감정결여의 경우는 괘씸각인에 의한 순종 업은 없음
		IF ABL:ARG:순종 <= 1 && !TALENT:ARG:감정결여
			ABL:ARG:순종 = 2
			TFLAG:각인순종변화 = 1
		ENDIF
	ENDIF
[SKIPEND]

A = CUP:ARG:공순 + CUP:ARG:굴복

;공순과 굴복에 의한 LvUP는 30번+취득한 괘씸각인의 레벨의 TFLAG를 준다
IF A >= 4500 && A < 7000 && MARK:ARG:괘씸각인 <= 0
	MARK:ARG:괘씸각인 = 1
	TFLAG:24 = 31
ELSEIF A >= 7000 && A < 10000 && MARK:ARG:괘씸각인 <= 1
	MARK:ARG:괘씸각인 = 2
	TFLAG:24 = 32
	;감정결여의 경우는 괘씸각인에 의한 순종 업은 없음
	IF !ABL:ARG:순종 && !TALENT:ARG:감정결여
		ABL:ARG:순종 = 1
		TFLAG:각인순종변화 = 1
	ENDIF
ELSEIF A >= 10000 && MARK:ARG:괘씸각인 <= 2
	MARK:ARG:괘씸각인 = 3
	TFLAG:24 = 33
	;감정결여의 경우는 괘씸각인에 의한 순종 업은 없음
	IF ABL:ARG:순종 <= 1 && !TALENT:ARG:감정결여
		ABL:ARG:순종 = 2
		TFLAG:각인순종변화 = 1
	ENDIF
ENDIF

;-------------------------------------------------
;반발각인
;-------------------------------------------------
;MASTER는 제외
SIF CFLAG:ARG:장난
	RETURN 0
SIF ARG == MASTER
	RETURN 0
SIF TFLAG:150
	RETURN 0
; 반발 이력을 남긴 경우, 조금만 반발 각인을 붙이기 어렵게 함
;같은 레벨 이하의 각인이 붙지 않으면 너무 강하므로 각인 레벨을 1단계 인하로
SIF MARK:ARG:반발취득이력
		각인취득치 = INRANGE(각인취득치, 500, 700) ? 0 # 각인취득치
SIF MARK:ARG:반발취득이력 > 1
		각인취득치 = INRANGE(각인취득치, 1001, 1300) ? 1000 # 각인취득치
SIF MARK:ARG:반발취득이력 > 2
		각인취득치 = INRANGE(각인취득치, 2001, 2500) ? 2000 # 각인취득치
각인취득치 = CUP:ARG:반감+CUP:ARG:불쾌
SELECTCASE 각인취득치
	CASE 500 TO 1000
		IF MARK:ARG:반발각인 < 1
			MARK:ARG:반발각인 = 1
			MARK:ARG:반발취득이력 = MAX(MARK:ARG:반발취득이력, 1)
			TFLAG:반발각인취득 = 1
			CFLAG:ARG:언짢음 = 1
			CFLAG:ARG:빡침 = 1
			TALENT:ARG:기분 = -1
		ENDIF
	CASE 1001 TO 2000
		IF MARK:ARG:반발각인 < 2
			MARK:ARG:반발각인 = 2
			TFLAG:반발각인취득 = 2
			MARK:ARG:반발취득이력 = MAX(MARK:ARG:반발취득이력, 2)
			CFLAG:ARG:언짢음 = 2
			CFLAG:ARG:빡침 = 1
			TALENT:ARG:기분 = -1
			IF MARK:ARG:반발취득이력 < 2 || 각인취득치 >= 1200
				;감정결여의 경우는 반발각인에 의한 순종 다운은 없음
				IF INRANGE(ABL:ARG:순종, 1,5) && !TALENT:ARG:감정결여
					ABL:순종 = MAX(ABL:순종 - 1, 0)
					ABL:친밀 = MAX(ABL:친밀 - 1, 0)
					TFLAG:각인순종변화 = 1
				ELSEIF ABL:ARG:순종 > 5 && !TALENT:ARG:감정결여
					ABL:순종 = MAX(ABL:순종 - 1, 0)
					TFLAG:각인순종변화 = 2
				ENDIF
			ENDIF
		ENDIF
	CASE IS > 2000
		IF MARK:ARG:반발각인 < 3
			MARK:ARG:반발각인 = 3
			TFLAG:반발각인취득 = 3
			MARK:ARG:반발취득이력 = MAX(MARK:ARG:반발취득이력, 3)
			CFLAG:ARG:언짢음 = 3
			CFLAG:ARG:빡침 = 1
			TALENT:ARG:기분 = -1
			IF MARK:ARG:반발취득이력 < 3 || 각인취득치 >= 2400
				;감정결여의 경우는 반발각인에 의한 순종 다운은 없음
				IF INRANGE(ABL:ARG:순종, 1,5) && !TALENT:ARG:감정결여
					ABL:순종 = MAX(ABL:순종 - 2, 0)
					ABL:친밀 = MAX(ABL:친밀 - 2, 0)
					TFLAG:각인순종변화 = 1
				ELSEIF ABL:ARG:순종 > 5 && !TALENT:ARG:감정결여
					ABL:순종 = MAX(ABL:순종 - 1, 0)
					ABL:친밀 = MAX(ABL:친밀 - 1, 0)
					TFLAG:각인순종변화 = 2
				ENDIF
			ENDIF
		ENDIF
ENDSELECT
ENDIF
;경고장 발행 체크
SIF !TALENT:ARG:연모 || RAND:(ABL:ARG:친밀 + 1) == 0 || MARK:ARG:반발각인 == 3 && FLAG:18
	CALL CARD_GOT_CHECK(ARG)
TFLAG:150 = 0

; 개인 임의수정, 각인 취득 판정시 스킵 일시정지 구문 추가
SIF TFLAG:고통각인취득 || TFLAG:시간각인취득 || TFLAG:쾌락각인취득 || TFLAG:24 || TFLAG:반발각인취득
	FORCEWAIT


code:
;FileName_TRACHECK.ERB ----------------------------- Rev1. 00
;신뢰도 상승 계산 처리
;CALL		USER
;ARG		0:캐릭터 NO
;RETURN		VOID
;COMMENT
;-----------------------------------------------------------
@TRUST_CALC(ARG)
#DIM 신뢰도상승치
DEBUGPRINTFORML 【TRUST_CALC】TRAST_CALC_START ,{ARG}
;COMABLE 범위외 억제
SIF !TFLAG:102
	RETURN 0
;대상이 MASTER 이외의때 기동
SIF !ARG
	RETURN 0

;실패 커맨드 플래그
SIF TFLAG:98 < 0
	TFLAG:99 ++

신뢰도상승치 = TFLAG:98

;친밀도에 맞추어 상승 수정
신뢰도상승치 += ABL:ARG:친밀 / 5
DEBUGPRINTFORML 【TRUST_CALC】초기치 TFLAG98={신뢰도상승치} 실패 플래그 {TFLAG:99}

;분노치에 의해 신뢰도 DOWN
SIF BASE:ARG:분노 > 0 && TFLAG:98 < 0
	신뢰도상승치 -= (BASE:ARG:분노 / 100)
DEBUGPRINTFORML 【TRUST_CALC】분노 변동 TFLAG98={신뢰도상승치}

;소질에 의한 증감

;반항적
SIF TALENT:ARG:태도 > 0 && !TALENT:ARG:연모
	신뢰도상승치 -= 1
;기분
신뢰도상승치 += TALENT:ARG:15
;솔직함
SIF TALENT:ARG:태도 < 0 && TFLAG:98 > 0
	신뢰도상승치 += 1
;명랑끼리
SIF TALENT:ARG:명랑／음침 > 0 && TALENT:MASTER:명랑／음침 > 0
	신뢰도상승치 += 1

;음침끼리
SIF TALENT:ARG:명랑／음침 < 0 && TALENT:MASTER:명랑／음침 < 0
	신뢰도상승치 += 1

;감정결여
SIF TALENT:ARG:감정결여
	신뢰도상승치 -= 1
;저항
SIF TALENT:ARG:저항
	신뢰도상승치 -= 2
;헌신적
SIF TALENT:ARG:헌신적 && TFLAG:98 > 0
	신뢰도상승치 += 1
;남성기피, 여성기피
SIF OTOKOGIRAI(ARG)
	신뢰도상승치 -= 1
;주인의 수수께끼의매력
SIF TALENT:MASTER:수수께끼의매력
	신뢰도상승치 += 1
SIF TALENT:ARG:스침 == 1
	신뢰도상승치 += 10
;연모
SIF TALENT:ARG:연모
	신뢰도상승치 += 2
;사모
SIF TALENT:ARG:사모
	신뢰도상승치 += 1
SIF TALENT:ARG:동물귀 && TALENT:ARG:동물귀 == TFLAG:귀보정
	신뢰도상승치 += 3
DEBUGPRINTFORML 【TRUST_CALC】소질 변동 TFLAG98={신뢰도상승치}

LOCAL = 0
LOCAL:1 = 0
;사랑
LOCAL += (50 - 250000 / (SOURCE:ARG:사랑 + 5000))
;달성감
LOCAL += (30 - 30000 / (SOURCE:ARG:달성 + 1000))
;공순
LOCAL += (20 - 20000 / (SOURCE:ARG:공순 + 1000))
LOCAL:2 += (20 - 20000 / (SOURCE:ARG:공순 + 1000))
;굴복
LOCAL += (15 - 15000 / (SOURCE:ARG:굴종 + 1000))
LOCAL:2 += (10 - 10000 / (SOURCE:ARG:굴종 + 1000))
;환락
LOCAL += (30 - 150000 / (SOURCE:ARG:환락 + 5000))
;정복
LOCAL += (30 - 90000 / (SOURCE:ARG:정복 + 3000))
;수동
LOCAL += (30 - 90000 / (SOURCE:ARG:수동 + 3000))
LOCAL:1 += LOCAL
;욕정
;LOCAL:2 += (30 - 90000 / (SOURCE:ARG:욕정 + 3000))
;쾌VACB
;LOCAL:2 += (20 - 200000 / (SOURCE:ARG:쾌V + 10000))
;LOCAL:2 += (20 - 200000 / (SOURCE:ARG:쾌A + 10000))
;LOCAL:2 += (10 - 100000 / (SOURCE:ARG:쾌C + 10000))
;LOCAL:2 += (10 - 100000 / (SOURCE:ARG:쾌B + 10000))
DEBUGPRINTFORML 【TRUST_CALC】UPPER 소스 변동 계산 {LOCAL}

;굴복
;LOCAL -= (50 - 50000 / (SOURCE:ARG:굴종 + 1000))
;LOCAL:2 += (50 - 50000 / (SOURCE:ARG:굴종 + 1000))
;공포
LOCAL -= (50 - 20000 / (SOURCE:ARG:공포 + 400)) * (ABL:ARG:순종 - 3) / 3
LOCAL:2 += (100 - 40000 / (SOURCE:ARG:공포 + 400)) * (ABL:ARG:순종 - 3) / 3
;고통
LOCAL -= (50 - 20000 / (SOURCE:ARG:고통 + 400)) * (ABL:ARG:마조끼 - 3) / 3
LOCAL:2 += (100 - 40000 / (SOURCE:ARG:고통 + 400)) * (ABL:ARG:마조끼 - 3) / 3
;불결
LOCAL -= (50 - 25000 / (SOURCE:ARG:불결 + 500)) * (2 - TALENT:ARG:악취내성) / 2
;울굴
LOCAL -= (50 - 15000 / (SOURCE:ARG:울굴 + 300))
;일탈
LOCAL -= (50 - 25000 / (SOURCE:ARG:일탈 + 500))
;반발
LOCAL -= (30 - 10000 / (SOURCE:ARG:반감 + 500))

DEBUGPRINTFORML 【TRUST_CALC】DOWNER 소스 변동 계산 {LOCAL:1 - LOCAL} ,{LOCAL}

신뢰도상승치 += LOCAL / 10
DEBUGPRINTFORML 【TRUST_CALC】소스 변동 TFLAG98={신뢰도상승치}

;기능이 낮으면 마이나스 보정
IF (TALENT:ARG:성적흥미 == -1 || TALENT:ARG:자존심 > 0) && 신뢰도상승치 > 0
	SIF ABL:ARG:교양 > ABL:MASTER:교양 || ABL:ARG:전투능력 > ABL:MASTER:전투능력
		신뢰도상승치 -= 1
ELSE
SIF (ABL:ARG:교양 > ABL:MASTER:교양 + 2 || ABL:ARG:전투능력 > ABL:MASTER:전투능력 + 2) && 신뢰도상승치 > 0
	신뢰도상승치 -= 1
ENDIF
IF GETBIT (CFLAG:ARG:인계, 1)
	신뢰도상승치 += 2
ELSEIF GETBIT (CFLAG:ARG:인계, 0)
	신뢰도상승치 += 1
ENDIF
;반발각인 보정
신뢰도상승치 -= (FLAG:4 + 1) * MARK:ARG:반발각인

;결과가 일정수미만의 경우
;IF MARK:반발각인 >= 3 && 신뢰도상승치 < -3
;	신뢰도상승치 = -3
;ELSEIF MARK:반발각인 == 2 && 신뢰도상승치 < -2
;	신뢰도상승치 = -2
;ELSEIF MARK:반발각인 == 1 && 신뢰도상승치 < -1
;	신뢰도상승치 = -1
;ENDIF

;동성은 신뢰도가 오르기 쉽고.
SIF HETEROSEX(ARG, PLAYER) && !OTOKOGIRAI(ARG)
	신뢰도상승치 += 1

;무관심한가 일선을넘지않음의 경우, 연모가 없으면 신뢰도가 오르기 어렵다
IF (TALENT:ARG:무관심 || TALENT:ARG:일선을넘지않음) && !TALENT:ARG:연모
	IF 신뢰도상승치 > 1
		신뢰도상승치 /= 2
	ELSEIF 신뢰도상승치 > 0
		신뢰도상승치 = 1
	ENDIF
ENDIF

IF FLAG:70
;	SIF 신뢰도상승치 > 0
;		신뢰도상승치 = 0
;	IF MARK:반발각인 >= 3
;		CFLAG:ARG:344 += TFLAG:98
;	ELSEIF MARK:반발각인 == 2
;		CFLAG:ARG:344 += TFLAG:98 / 2
;	ELSEIF MARK:반발각인 == 1
;		CFLAG:ARG:344 += TFLAG:98 / 3
;	ELSE
;		CFLAG:ARG:344 += TFLAG:98 / 4
;	ENDIF
ELSE
[SKIPSTART]
;불성실 때마다 밤페널티
SELECTCASE TCVAR:ARG:불성실도
	CASE IS <= 10
		 신뢰도상승치 ++
	CASE 11 TO 20
	CASE 21 TO 50
		신뢰도상승치 -= 1
	CASE 51 TO 100
		신뢰도상승치 -= 3
	CASE 101 TO 300
		신뢰도상승치 -= 5
	CASEELSE
		신뢰도상승치 -= 7
ENDSELECT
[SKIPEND]
IF CFLAG:ARG:몸목적
	SIF 신뢰도상승치 > 0
		신뢰도상승치 /= 2
ENDIF

;이성과 무드로 수정
IF BASE:ARG:무드 == MAXBASE:ARG:무드
	신뢰도상승치 += 2
	;무드가 전개라면 신뢰도 마이나스 없음
	SIF 신뢰도상승치 < 0
		신뢰도상승치 = 0
ELSEIF BASE:ARG:무드 >= (MAXBASE:ARG:무드 / 5 * 4) || BASE:ARG:이성 <= (MAXBASE:ARG:이성 / 2)
	신뢰도상승치 += 1
	;무드가4/5이상 또는 이성이1/2이하때, 신뢰도 마이나스가 반
	SIF 신뢰도상승치 < 0
		신뢰도상승치 += ABS(신뢰도상승치) / 2
ENDIF
DEBUGPRINTFORML 【TRUST_CALC】무드 변동 {신뢰도상승치}

SIF TFLAG:호감도보너스 && 신뢰도상승치 > 0
	;신뢰도상승치 = 신뢰도상승치 * TFLAG:호감도보너스
	신뢰도상승치 = 신뢰도상승치 * (100 + TFLAG:호감도보너스) / 100

;신뢰도 상승이 낮은 경우, 거점 거주자이며 모브가 같은 장소에 있으면 보너스
SIF 신뢰도상승치 < 3 && WITH_MOB() && MAP_거주자(MAIN_MAP, ARG)
	신뢰도상승치 ++
;실패 플래그가 서지 않으면 신뢰는 내리지 않는다
SIF !TFLAG:99 && 신뢰도상승치 < 0
	신뢰도상승치 = 0
SIF TFLAG:99 && 신뢰도상승치 > 0
	신뢰도상승치 = 0
;소프트 우후후시는 플러스 없음
SIF SELECTCOM >= 313 && SELECTCOM <= 330 && 신뢰도상승치 > 0
	신뢰도상승치 = 0
;아무것도 하지 않는을 사용했을 경우는 상한을 0으로 한다
SIF 신뢰도상승치 > 0 && SELECTCOM == 21
	신뢰도상승치 = 0
;조수 조교중
SIF ASSIPLAY == 1
	신뢰도상승치 = 0
;수면중은 상승치 0
SIF CFLAG:ARG:수면 || TCVAR:TARGET:만취
	신뢰도상승치 = 0

;신뢰도 상승의 하한 설정. (반발각인의 레벨에 기분의 수정을 더한다)×-5(이)가 하한)
신뢰도상승치 = MAX(신뢰도상승치 , MAX(MARK:ARG:반발각인 - TALENT:ARG:15 , 0) * -5)
신뢰도상승치=신뢰도상승치*(1+FLAG:호감신뢰상승률)
DEBUGPRINTFORML 【TRUST_CALC】신뢰도 변동 {신뢰도상승치}
SIF LOCAL:1
	CFLAG:ARG:신뢰도 += 신뢰도상승치
;신뢰도는 0 이하에도 100만 이상이나 되지 않는다
CFLAG:ARG:신뢰도 = LIMIT(CFLAG:ARG:신뢰도, 0, 999999)
LOCAL:1 = 0
FOR LOCAL, 0,100
	SIF SOURCE:ARG:LOCAL
		LOCAL:1 ++
NEXT
SIF !LOCAL:1
	신뢰도상승치 = 0

;신뢰도 표시
SIF 신뢰도상승치
PRINTFORML \@(신뢰도상승치 >= 0) ?  신뢰도＋{신뢰도상승치} # 신뢰도-{-신뢰도상승치}\@
ENDIF
TFLAG:98 = 0
TFLAG:99 = 0
신뢰도상승치 = 0


code:
;FileName_TRACHECK.ERB ----------------------------- Rev1. 00
;호감도 상승 계산 처리
;CALL		USER
;ARG		0:캐릭터 NO
;RETURN		VOID
;COMMENT
;-----------------------------------------------------------
@FAVOR_CALC(ARG)
#DIM 분노상승치
#DIM 무드상승치
#DIM 신뢰도캡보정
#DIM 이성감소치
SIF FLAG:70 == 1 && CFLAG:TARGET:344 == 0
	RETURN

;COMABLE 범위외 억제
SIF !TFLAG:102
	RETURN 0
;대상이 MASTER 이외의때 기동
SIF !ARG
	RETURN 0
;주인(당신)과 대상의 능력을 계산
;ABL:순종을 본다
TFLAG:97 += (ABL:ARG:순종 + ABL:ARG:친밀) / 2

;주인과 대상이 동성
IF HETEROSEX(TARGET, PLAYER)
	;남자라면 호모끼, 여자라면 레즈끼를 본다
	IF ABL:ARG:(HETEROSEX(TARGET, PLAYER) + 16) == 0
		TFLAG:97 -= 1
	ELSEIF ABL:ARG:(HETEROSEX(TARGET, PLAYER) + 16) < 5
		TFLAG:97 += 0
	ELSEIF ABL:ARG:(HETEROSEX(TARGET, PLAYER) + 16) < 9
		TFLAG:97 += 1
	ELSE
		TFLAG:97 += 2
	ENDIF
	;남자라면 BL중독, 여자라면 레즈중독을 본다
	IF ABL:ARG:(HETEROSEX(TARGET, PLAYER) + 31) > 9
		TFLAG:97 += 2
	ELSEIF ABL:ARG:(HETEROSEX(TARGET, PLAYER) + 31) > 4
		TFLAG:97 += 1
	ENDIF
	;양도
	SIF TALENT:ARG:성별기호 == -1
		TFLAG:97 += 1
ENDIF
;소질에 의한 증감
;반항적
SIF TALENT:ARG:태도 > 0
	TFLAG:97 -= 2
;자존심 높음
SIF TALENT:ARG:자존심 > 0
	TFLAG:97 -= 1
;솔직함
SIF TALENT:ARG:태도 < 0 && TFLAG:97 > 0
	TFLAG:97 += 1
;자제심
SIF TALENT:ARG:자제심
	TFLAG:97 -= 1
;감정결여
SIF TALENT:ARG:감정결여
	TFLAG:97 -= 1
;저항
SIF TALENT:ARG:저항
	TFLAG:97 -= 1
;헌신적
SIF TALENT:ARG:헌신적 && TFLAG:97 > 0
	TFLAG:97 += 1
;명랑끼리
SIF TALENT:ARG:명랑／음침 > 0 && TALENT:MASTER:명랑／음침 > 0
	TFLAG:97 += 1

;음침끼리
SIF TALENT:ARG:명랑／음침 < 0 && TALENT:MASTER:명랑／음침 < 0
	TFLAG:97 += 1
;쾌감에 솔직
IF TALENT:ARG:쾌감응답 > 0
	TFLAG:97 += 1
;쾌감에 부정
ELSEIF TALENT:ARG:쾌감응답 < 0
	TFLAG:97 -= 1
ENDIF
;기분
TFLAG:97 += TALENT:ARG:15
;남성기피
IF OTOKOGIRAI(ARG)
	TFLAG:97 -= 3
	IF GETBIT (CFLAG:ARG:인계, 1)
		TFLAG:97 += 2
	ELSEIF GETBIT (CFLAG:ARG:인계, 0)
		TFLAG:97 += 1
	ENDIF
ENDIF
;연모
IF TALENT:ARG:연모
	TFLAG:97 += 2
;춘약
ELSEIF TCVAR:ARG:춘약
	TFLAG:97 ++
	SIF CFLAG:ARG:2 < 1000
		TFLAG:97 ++
ENDIF
;仙香玉兎
IF IsInsenceValid(ARG)
	TFLAG:97 ++
	SIF CFLAG:ARG:2 < 1000
		TFLAG:97 ++
ENDIF
;ガシングガーデン
SIF IsGardenValid(ARG)
	TFLAG:97 ++

SIF TALENT:ARG:동물귀 && TALENT:ARG:동물귀 == TFLAG:귀보정
	TFLAG:97 += 3
;사모
SIF TALENT:ARG:사모
	TFLAG:97 += 1
;주인의 용모
;TFLAG:97 += TALENT:MASTER:용모
;주인의 매력
SIF TALENT:MASTER:매력
	TFLAG:97 += 1
;주인의 매혹
SIF TALENT:MASTER:매혹
	TFLAG:97 += 1
;주인의 수수께끼의매력
SIF TALENT:MASTER:수수께끼의매력
	TFLAG:97 += 1
;주인의 인기
SIF TALENT:MASTER:인기
	TFLAG:97 += 1

IF GETBIT (CFLAG:ARG:인계, 1)
	TFLAG:97 += 2
ELSEIF GETBIT (CFLAG:ARG:인계, 0)
	TFLAG:97 += 1
ENDIF

LOCAL = 0
LOCAL:1 = 0
;각종 쾌감
;50상한, 30000으로+300 이 수치가 크면 SOURCE:ARG의 증가에 대해서 둔하게 된다
LOCAL += (100 - 30000 / (SOURCE:ARG:쾌C + SOURCE:ARG:쾌V + SOURCE:ARG:쾌A + SOURCE:ARG:쾌B + 300)) * (1 + TALENT:ARG:쾌감응답)
;사랑
LOCAL += (100 - 100000 / (SOURCE:ARG:사랑 + 1000))
LOCAL:1 += (100 - 100000 / (SOURCE:ARG:사랑 + 1000))
;욕정
LOCAL += (20 - 20000 / (SOURCE:ARG:욕정 + 1000))
LOCAL:1 += (100 - 100000 / (SOURCE:ARG:욕정 + 1000))
;달성감
LOCAL += (30 - 30000 / (SOURCE:ARG:달성 + 1000))
;공순
LOCAL += (20 - 20000 / (SOURCE:ARG:공순 + 1000))
;굴복
LOCAL += (20 - 20000 / (SOURCE:ARG:굴종 + 1000))
;노출
LOCAL += (30 - 30000 / (SOURCE:ARG:노출 + 1000)) * (ABL:ARG:노출벽 - 3) / 3

;고통
LOCAL += (50 - 20000 / (SOURCE:ARG:고통 + 400)) * (ABL:ARG:마조끼 - 3) / 3
;환락
LOCAL += (50 - 10000 / (SOURCE:ARG:환락 + 2000))
;정복
LOCAL += (30 - 90000 / (SOURCE:ARG:정복 + 3000))
;수동
LOCAL += (30 - 90000 / (SOURCE:ARG:수동 + 3000))
;공포
LOCAL -= (50 - 20000 / (SOURCE:ARG:공포 + 400)) * (ABL:ARG:순종 - 3) / 3
;불결
LOCAL -= (50 - 25000 / (SOURCE:ARG:불결 + 500)) * (2 - TALENT:ARG:악취내성) / 2
;울굴
LOCAL -= (50 - 15000 / (SOURCE:ARG:울굴 + 300))
;일탈
LOCAL -= (50 - 25000 / (SOURCE:ARG:일탈 + 500))
;반발
LOCAL -= (50 - 10000 / (SOURCE:ARG:반감 + 200))

;호감도 상승분에 계산치 추가
TFLAG:97 += LOCAL / 10

;일상시 ON
IF TFLAG:102 != 2
	;무드에 호감도 상승분 추가
	무드상승치 = MIN(ABL:ARG:친밀 + 1,10) * LOCAL / 3
	SIF FLAG:데이트상대 == ARG
		TFLAG:무드보너스 *= 2
	무드상승치 = GET_REVISION3(무드상승치, 50 + TFLAG:무드보너스, 100 + TFLAG:무드보너스, 200 + TFLAG:무드보너스)

	;억제 플래그가 서 있다
	IF TFLAG:무드상승억제
		무드상승치 /= 2
		SIF BASE:ARG:무드 > MAXBASE:ARG:무드 / 2
			무드상승치 = 0
	ENDIF
	;모브가 있으면(자) 상승량 반감
	SIF WITH_MOB()
		무드상승치 /= 2
	BASE:ARG:무드 += MIN(무드상승치, 200)
	BASE:ARG:무드 = MIN(BASE:ARG:무드 , MAXBASE:ARG:무드)
	;같은 방에 파루파루가 있는 상태로 다른 애와 무드를 올리면 분노가 모인다
	IF ARG != 60 && CFLAG:ARG:현재위치 == CFLAG:60:현재위치
		IF TALENT:60:연모
			BASE:60:분노 += MIN(ABL:60:친밀 + 1,10) * LOCAL / 3
		ELSEIF TALENT:60:사모
			BASE:60:분노 += MIN(ABL:60:친밀 + 1,10) * LOCAL / 5
		ELSEIF CFLAG:60:2 > 500
			BASE:60:분노 += MIN(ABL:60:친밀 + 1,10) * LOCAL / 8
		ENDIF
	ENDIF
	;이성을 쾌감도분감소
	이성감소치 = LOCAL:1 * (GETPALAMLV(PALAM:ARG:욕정, 10) + 1) / 10 * (CFLAG:ARG:욕구불만도 + 1000) / 2000
	IF TFLAG:이성깎기
		이성감소치 *= 4
		SIF CFLAG:ARG:몸목적
			이성감소치 *= 2
	ENDIF
	BASE:ARG:이성 = MAX(BASE:ARG:이성 - 이성감소치, 0 )
	분노상승치 = 0
	;일상 ON 또한 반감 소스유로 분노 가격인상승
	SIF SOURCE:ARG:반감 > 0 && TFLAG:102 == 1
		분노상승치 = SOURCE:ARG:반감 * 3 * (2000 - BASE:ARG:무드) * (500 + BASE:ARG:이성) / ((2000 + BASE:ARG:무드) * (500 + MAXBASE:ARG:이성))
	;바이브 팬티나 촉수 속옷을 입고 있는 때는 분노치가 상승하지 않는 분노 방지
	SIF SOURCE:ARG:반감 > 0 && TFLAG:102 == 1 && (EQUIP:ARG:하반신속옷2 == 28 || EQUIP:ARG:하반신속옷2 == 29 || EQUIP:ARG:하반신속옷2 == 30)
		분노상승치 = 0
	;정조대를 입고 있는 때는 분노상승치가 증가
	SIF SOURCE:ARG:반감 > 0 && TFLAG:102 == 1 && EQUIP:ARG:하반신속옷2 == 27
		분노상승치 = 분노상승치 * 4 / 3
	;기분으로 분노상승치가 변화
	IF TALENT:ARG:15 == 1
		분노상승치 = 분노상승치 * 7 / 10
	ELSEIF TALENT:ARG:15 == -1
		분노상승치 = 분노상승치 * 3 / 2
	ENDIF
	SIF TCVAR:ARG:수면제강도 == 1 && CFLAG:ARG:장난
		분노상승치 = 분노상승치 / 2
	SIF TCVAR:ARG:수면제강도 == 2 && CFLAG:ARG:장난
		분노상승치 = 분노상승치 / 5
	SIF TCVAR:ARG:수면제강도 == 3 && CFLAG:ARG:장난
		분노상승치 = 0
	BASE:ARG:분노 += 분노상승치
	;분노가 매우 기뻐해 분노 플래그 ON
	SIF BASE:ARG:분노 > MAXBASE:ARG:분노
		CALL GET_ANGRY(ARG)
	BASE:ARG:분노 = MIN(BASE:ARG:분노 , MAXBASE:ARG:분노 )
	;언짢음(반발각인취득시)은 항상 분노
	IF CFLAG:ARG:언짢음
		BASE:ARG:분노 = MAXBASE:ARG:분노
		CFLAG:ARG:빡침 = 1
	ENDIF
	;화내면(자) 눈을 뜬다
	IF CFLAG:ARG:빡침 && CFLAG:ARG:수면
		CALL CHARA_AWAKE(ARG, 1)
		IF CFLAG:ARG:현재위치 == CFLAG:MASTER:현재위치 && CFLAG:ARG:현재위치 == CFLAG:ARG:311
			IF IS_LOVER(ARG)
				IF CFLAG:ARG:쇠약
					PRINTFORML %ARG는% 내방자가 %CALLNAME:MASTER%인 것 확인하면(자) 지쳐 자 버렸다…
				ELSE
					PRINTFORML %ARG는% 내방자가 %CALLNAME:MASTER%인 것 확인하면(자) 안도의 표정을 보였다…
				ENDIF
			ELSE
				IF CFLAG:ARG:쇠약
					PRINTFORML %ARG는% 지쳐 자 버렸다…
				ELSE
					PRINTFORML %CALLNAME:ARG%에 내쫓아졌다…
				ENDIF
				CALL GETOUT(MASTER)
			ENDIF
		ENDIF
	ENDIF
	;경고장 발행 체크
	SIF (CFLAG:ARG:빡침 == 1) && RAND:((ABL:ARG:친밀 + 1) * 3) == 0 && !CFLAG:ARG:수면 && FLAG:18
		CALL CARD_GOT_CHECK(ARG)
ENDIF

;반발각인 보정
TFLAG:97 -= (FLAG:4 + 1) * MARK:반발각인

;결과가 일정수미만의 경우
IF MARK:반발각인 >= 3 && TFLAG:97 < -3
	TFLAG:97 = -3
ELSEIF MARK:반발각인 == 2 && TFLAG:97 < -2
	TFLAG:97 = -2
ELSEIF MARK:반발각인 == 1 && TFLAG:97 < -1
	TFLAG:97 = -1
ELSEIF TFLAG:97 < 0
	TFLAG:97 = 0
ENDIF

;무관심의 경우, 연모가 없으면 호감도가 오르기 어렵다
IF TALENT:ARG:무관심 && !TALENT:ARG:연모
	IF TFLAG:97 > 1
		TFLAG:97 /= 2
	ELSEIF TFLAG:97 > 0
		TFLAG:97 = 1
	ENDIF
	SIF ABL:ARG:새드끼 >= 4
		TFLAG:97 += 1
ENDIF
;걸레라면 친밀 5로부터 올리기 어려워진다
IF TALENT:ARG:걸레
	TFLAG:97 += 2
	SIF ABL:ARG:친밀 >= 5
		TFLAG:97 /= 2
ENDIF
[SKIPSTART]
;불성실 때마다 밤패널티
SELECTCASE TCVAR:ARG:불성실도
	CASE IS <= 10
			TFLAG:97 ++
	CASE 11 TO 20
	CASE 21 TO 50
		TFLAG:97 -= 1
	CASE 51 TO 100
		TFLAG:97 -= 3
	CASE 101 TO 300
		TFLAG:97 -= 5
	CASEELSE
		TFLAG:97 -= 10
ENDSELECT
[SKIPEND]
;조수 조교중
SIF ASSIPLAY == 1
	TFLAG:97 = 0
;억지로
SIF TCVAR:ARG:억지로
	TFLAG:97 = MIN(0, TFLAG:97 * -3)
;대상이 자고 있으면(자) 상승치 0
SIF !SHIRAHU(ARG)
	TFLAG:97 = 0
;호감도가 신뢰도의 4배를 넘으면(자) 상승치가 저하 다만 데이트중은 예외
IF CFLAG:ARG:2 > 500 && CFLAG:ARG:호감도 > CFLAG:ARG:신뢰도 * 4
	신뢰도캡보정 = CFLAG:ARG:호감도
	SIF FLAG:데이트상대 == ARG
		신뢰도캡보정 /= 2
	IF TALENT:ARG:연모
		TFLAG:97 = GET_REVISION(CFLAG:ARG:신뢰도 * 40, TFLAG:97, 신뢰도캡보정)
	ELSEIF TALENT:ARG:사모
		TFLAG:97 = GET_REVISION(CFLAG:ARG:신뢰도 * 30, TFLAG:97, 신뢰도캡보정)
	ELSE
		TFLAG:97 = GET_REVISION(CFLAG:ARG:신뢰도 * 20, TFLAG:97, 신뢰도캡보정)
	ENDIF
ENDIF
;동시에 절정
SIF SYNCED_ORGASM(ARG)
	TFLAG:97 = TFLAG:97 * 3 / 2
;상대 주도 우후후
SIF CFLAG:우후후 == 2
	TFLAG:97 = TFLAG:97 * 6 / 5
;스침
SIF TALENT:ARG:스침 == 1
	CALL YUKIZURI_BONUS(ARG)
;호감도마이너스
SIF TFLAG:호감도마이너스 && TFLAG:97 > 0
	TFLAG:97 = TFLAG:97 * -1 *TFLAG:호감도마이너스
SIF TFLAG:호감도보너스 && TFLAG:97 > 0
	;TFLAG:97 = TFLAG:97 * TFLAG:호감도보너스
	TFLAG:97 = TFLAG:97 * (100 + TFLAG:호감도보너스) / 100

;호감도 상승이 낮은 경우, 거점 거주자이며 모브가 같은 장소에 있으면 보너스
SIF TFLAG:97 < 3 && WITH_MOB() && MAP_거주자(MAIN_MAP, ARG)
	TFLAG:97 ++
;커맨드 실패했을 경우 호감 도는 오르지 않는다
SIF TFLAG:98 < 0 || TFLAG:호감도안들어감
	TFLAG:97 = MIN(0, TFLAG:97)

SIF TFLAG:3 == 300 && TFLAG:193 < 0
	TFLAG:97 = 0

;만취간 들키고 때는 문답무용으로 호감도 대폭 저하
SIF TCVAR:ARG:만취간들킴
	TFLAG:97 -= MAX(30 - ABL:ARG:친밀, 0) * 3

;KR판 오리지날 호감신뢰상승률
TFLAG:97=TFLAG:97*(1+FLAG:호감신뢰상승률+TALENT:수수께끼의매력*2)

;호감 때마다 상승치를 가산
CFLAG:ARG:2 += TFLAG:97

;호감 도는 0 이하에도 100만 이상이나 되지 않는다
CFLAG:ARG:2 = LIMIT(CFLAG:ARG:2, 0, 1000000)
LOCAL:1 = 0
FOR LOCAL, 0,100
	SIF SOURCE:ARG:LOCAL
		LOCAL:1 ++
NEXT
SIF !LOCAL:1
	TFLAG:97 = 0

;호감도 표시
SIF TFLAG:97
	PRINTFORML \@(TFLAG:97 >= 0) ? 호감도＋{TFLAG:97} # 호감도-{-TFLAG:97}\@
;호감도상승치 합산
FLAG:12 += TFLAG:97

@YUKIZURI_BONUS(ARG)
TFLAG:97 *= 10
SIF CFLAG:ARG:몸목적
	TFLAG:97 *= 2
LOCAL = GETPALAMLV(PALAM:ARG:호의, 10) - 1
CALL YUKIZURI_BONUS_1(ARG, "친밀", LOCAL)

LOCAL = GETPALAMLV(PALAM:ARG:습득,3)
CALL YUKIZURI_BONUS_1(ARG, "기교", LOCAL)

LOCAL = MAX(GETPALAMLV(PALAM:ARG:공순,3), GETPALAMLV(PALAM:ARG:굴복,3))
CALL YUKIZURI_BONUS_1(ARG, "순종", LOCAL)

LOCAL = GETPALAMLV(PALAM:ARG:공순,3) - 1
CALL YUKIZURI_BONUS_1(ARG, "봉사정신", LOCAL)

@YUKIZURI_BONUS_1(C_ID, CHK_ABL, ARG)
#DIM C_ID
#DIMS CHK_ABL

SIF !ARG
	RETURN
SIF ABL:C_ID:CHK_ABL >= ARG
	RETURN
LOCAL = FINDELEMENT(ABLNAME, CHK_ABL)
CALL COLORMESSAGE(@"%조사처리(ABLNAME:LOCAL,"가")% {ARG} 올랐다", C_YELLOW,2,1)
ABL:C_ID:CHK_ABL = ARG


code:
@밀어넘어뜨려지고
#DIM 인기척
#DIM 밀어넘어뜨려짐판정
#DIM 껴안기값
#DIM 밀어넘어뜨려짐값
#DIM 껴안기우선
#DIM C_ID

IF FLAG:덮치기
	밀어넘어뜨려짐판정 = 0
	밀어넘어뜨려짐판정 += ABL:TARGET:전투능력 - ABL:MASTER:전투능력
	밀어넘어뜨려짐판정 += TALENT:TARGET:체형 - TALENT:MASTER:체형
	IF CFLAG:TARGET:수줍음 > 0
		밀어넘어뜨려짐판정 += 1
	ELSEIF CFLAG:TARGET:수줍음 < 0
		밀어넘어뜨려짐판정 -= 1
	ENDIF
	IF CFLAG:TARGET:유혹 > 0
		밀어넘어뜨려짐판정 += 1
	ELSEIF CFLAG:TARGET:유혹 < 0
		밀어넘어뜨려짐판정 -= 1
	ENDIF
	IF CFLAG:TARGET:적극성 > 0
		밀어넘어뜨려짐판정 += 1
	ELSEIF CFLAG:TARGET:적극성 < 0
		밀어넘어뜨려짐판정 -= 1
	ENDIF
	IF CFLAG:TARGET:질 > 0
		밀어넘어뜨려짐판정 += 1
	ELSEIF CFLAG:TARGET:질 < 0
		밀어넘어뜨려짐판정 -= 1
	ENDIF
	IF GROUPMATCH(CFLAG:TARGET:성격경향,1,3)
		밀어넘어뜨려짐판정 += 1
	ELSEIF GROUPMATCH(CFLAG:TARGET:성격경향,2,4)
		밀어넘어뜨려짐판정 -= 1
	ENDIF
	;밀어넘어뜨려짐판정 += ABL:TARGET:새드끼 + ABL:MASTER:마조끼
	;밀어넘어뜨려짐판정 -= TALENT:MASTER:담력 + TALENT:MASTER:태도 + TALENT:MASTER:응답 + TALENT:MASTER:자존심 + TALENT:MASTER:일선을넘지않음 + TALENT:MASTER:저항 + TALENT:MASTER:정조
	;밀어넘어뜨려짐판정 += (CFLAG:TARGET:약점잡음 * 2) + (TALENT:TARGET:새드 * 3) + (TALENT:MASTER:마조 * 3) + (TCVAR:TARGET:발정 * 3)
	IF CFLAG:TARGET:욕구불만도 < 600
		밀어넘어뜨려짐판정 = 0
	ELSEIF CFLAG:TARGET:욕구불만도 < 800
		밀어넘어뜨려짐판정 += 1
	ELSEIF CFLAG:TARGET:욕구불만도 < 1000
		밀어넘어뜨려짐판정 += 2
	ELSE
		밀어넘어뜨려짐판정 += 3
	ENDIF
ENDIF

IF CFLAG:우후후 == 1
	TFLAG:102 = 2
ELSEIF CFLAG:우후후 == 2
	TFLAG:102 = 3
ELSEIF TFLAG:102 == 3 && CFLAG:MASTER:우후후 == 2
	;이미 처리되었으므로 스킵
ELSE
	TFLAG:102 = 1
ENDIF

IF !(TFLAG:102 == 3 && CFLAG:MASTER:우후후 == 2)
	TFLAG:껴안기모드 = 0
ENDIF

;껴안김 or 밀어 넘어뜨려짐의 처리
;데이트 중에는 발생하지 않는다
IF !CHK_DATENOW(CFLAG:TARGET:데이트중)
	;껴안김과 밀어 넘어뜨려짐, 어느 쪽이 우선해 실행되는지를 판정. 껴안기는 것은 여자끼리 한정
	IF (TALENT:MASTER:성별 & 1) && (TALENT:성별 & 1)
		껴안기값 = 안아다하고판정(TARGET)
	ELSE
		껴안기값 = 0
	ENDIF
	밀어넘어뜨려짐값 = ENDURE(TARGET)

	껴안기우선 = (껴안기값 >= 1 && 껴안기값 > 밀어넘어뜨려짐값 + 40 && !(밀어넘어뜨려짐값 >= 600 && BASE:이성 <= 300))

	;껴안기가 우선인 경우
	IF 껴안기우선
		IF !CFLAG:우후후 && TFLAG:102 != 3 && 껴안기값 >= RAND:250 + 250 && TCVAR:껴안기금지시간 <= 0
			;실외에서는 껴안지 않는다
			IF (!HITOKE() && INROOM(CFLAG:MASTER:현재위치) && AT_HOME(MASTER)) || CFLAG:MASTER:현재위치 == OMANEKIBEYA()
				;같은 방에 있는 게 1명만인 경우(껴안는 사람은 여러 명 불가)
				IF GET_TARGETNUM() == 1
					PRINTL
					PRINTFORMW %마스터는% %CALLNAME%에게 안겨졌다!
					CALL KOJO_MESSAGE_SEND("EVENT",10,TARGET,1,1)
					TFLAG:껴안기해방포인트 = 2 + RAND:(MIN(ABL:욕망, 10) / 2 + 2) + MIN(CFLAG:욕구불만도, 1000) / 200
					CFLAG:우후후 = 2
					CFLAG:MASTER:우후후 = 2
					TFLAG:102 = 3
					TFLAG:껴안기모드 = 1
					IF !TCVAR:발정 && !TCVAR:미약 && !GETBIT(CFLAG:기정사실 , 1)
						SETBIT CFLAG:기정사실 , 1
						CALL COLORMESSAGE(@"%CALLNAME%의 합의를 얻었다",C_YELLOW,2)
					ELSE
						CFLAG:대주고말았다 = 1
					ENDIF
				ENDIF
			ENDIF
		ENDIF
	;밀어 넘어뜨려짐이 우선인 경우
	ELSE
		IF ((!CFLAG:우후후 && TFLAG:102 != 3) || (TFLAG:102 == 3 && TFLAG:껴안기모드)) && 밀어넘어뜨려짐값 >= RAND:300 + 300
			;실외에서는 밀어넘어뜨려지지 않는다.
			IF (!HITOKE() && INROOM(CFLAG:MASTER:현재위치) && AT_HOME(MASTER)) || CFLAG:MASTER:현재위치 == OMANEKIBEYA()
				;같은 방에 있는 게 1명만인 경우 또는 복수인에게 밀렸을 때
				IF GET_TARGETNUM() == 1 || 복수밀어넘어뜨려지고() == 1
					;押し倒されない効果
					IF FLAG:덮치기 && 밀어넘어뜨려짐판정 <= 0
						PRINTL
						PRINTFORML 갑자기 %CALLNAME:TARGET%에게 안겨졌다!
						PRINTFORM %타겟은%
						IF PALAM:TARGET:욕정 >= PALAMLV:10
							PRINT 반짝거리는
						ELSE
							PRINT 물기를 띤
						ENDIF
						PRINTFORML 눈동자로 바라봤다
					;딜레이 중에는 덮치지 않는다
					ELSEIF CFLAG:TARGET:데일리 > 0

					ELSE
						PRINTL
						PRINTFORMW %마스터는% 밀어 넘어뜨려졌다!
						IF !UFUFU_START_TIME
							UFUFU_START_TIME = TIME
							UFUFU_START_DAY = DAY
						ENDIF
						;CALL KOJO_EVENT(12, TARGET)
						CALL KOJO_MESSAGE_SEND("EVENT",10,TARGET,1,0)
						FOR LOCAL, 1, GET_TARGETNUM() + 1
							C_ID = TARGET:LOCAL
							CFLAG:C_ID:우후후 = 2
							IF !TCVAR:C_ID:발정 && !TCVAR:C_ID:미약 && !GETBIT (CFLAG:C_ID:기정사실 , 1)
								SETBIT CFLAG:C_ID:기정사실 , 1
								CALL COLORMESSAGE(@"%CALLNAME:C_ID%의 합의를 얻었다",C_YELLOW,2)
							ELSE
								CFLAG:C_ID:대주고말았다 = 1
							ENDIF
						NEXT
						CFLAG:MASTER:우후후 = 2
						TFLAG:102 = 3
						TFLAG:껴안기모드 = 0
						TFLAG:덮치기개시시간 = TIME + DAY * 1440
					ENDIF
				ENDIF
				IF INROOM(CFLAG:MASTER:현재위치) == 2
					CALL DATUI_SHOES(MASTER,0)
					CALL DATUI_SHOES(TARGET,0)
				ENDIF
			;실외(여관과 러브호텔 제외)에서 밀어 넘어뜨려지는 조건을 충족하는 경우
			ELSEIF CFLAG:MASTER:현재위치 != 260 && CFLAG:MASTER:현재위치 != 950
				PRINTL
				SELECTCASE CFLAG:TARGET:욕구불만도
					CASE IS < 600
						PRINTFORML %타겟이% 소매를 잡아당겼다
						PRINTFORMW 어딘가 남의 눈에 띄지 않는 장소에 가고 싶은 모양이다
					CASE IS < 800
						PRINTFORMW %타겟은% %CALLNAME:MASTER%에게 달라붙어서, 눈을 치뜨고 바라봤다…
					CASE IS < 1000
						PRINTFORML %타겟이% %CALLNAME:MASTER%에게 달라붙어서, \@ BATHROOM(CFLAG:현재위치) ? 노출된 # 옷 위로 \@ 고간을 살짝 쓰다듬었다…
						PRINTFORMW %타겟은% 욕정을 시선으로 호소하고 있다
						IF TCVAR:TARGET:유혹시각 != TIME
							PALAM:TARGET:윤활 += 500
							TCVAR:TARGET:유혹시각 = TIME
						ENDIF
					CASEELSE
						PRINTFORML %타겟은% 상기된 얼굴로 %CALLNAME:MASTER%에게 아양 떨며 기대고, %CALLNAME:MASTER%의 손을 잡고 자신의 흠뻑 젖어버린 비부에 이끌었다…
						PRINTFORMW %타겟은% 지금 당장이라도 %CALLNAME:MASTER%에게 안기고 싶어 견딜 수 없는 것 같다
						IF TCVAR:TARGET:유혹시각 != TIME
							PALAM:TARGET:윤활 += 2000
							TCVAR:TARGET:유혹시각 = TIME
						ENDIF
				ENDSELECT
				CALL KOJO_MESSAGE_SEND("EVENT",10,TARGET,4, CFLAG:TARGET:욕구불만도)
			ENDIF
		ENDIF
	ENDIF
ENDIF

;복수 밀어 넘어뜨려짐 판정
@복수밀어넘어뜨려지고()
#FUNCTION
#DIM 판정치
#DIM 캐릭터별판정치
#DIM 궁합보정
#DIM NUM_TARGET
#DIM C_ID

판정치 = 0
NUM_TARGET = GET_TARGETNUM()
SIF NUM_TARGET <= 1
	RETURNF 0
FOR LOCAL, 1, NUM_TARGET + 1
	C_ID = TARGET:LOCAL
	SIF CFLAG:C_ID:동실 != 1
		CONTINUE
	SIF ENDURE(C_ID) < 450
		RETURNF 0
	;판정을 통과하면 캐릭터별 판정치에 넣을 수 있다
	캐릭터별판정치 = ENDURE(C_ID)
	DEBUGPRINTFORML 보정 전：({C_ID}) {ENDURE(C_ID)}
	;여기에서 캐릭터별 판정치를 보정해 나간다
	;레즈끼에서의 보정(3 미만에서는 패널티)
	SIF ABL:(C_ID):레즈끼 == 0
		TIMES 캐릭터별판정치, 0
	SIF ABL:(C_ID):레즈끼 == 1
		TIMES 캐릭터별판정치, 0.50
	SIF ABL:(C_ID):레즈끼 == 2
		TIMES 캐릭터별판정치, 0.80
	;3 이상에서는 (1+0.1*(레즈끼 - 3))배
	SIF ABL:(C_ID):레즈끼 >= 3
		캐릭터별판정치 = 캐릭터별판정치 * (100 + 10 * (ABL:(C_ID):레즈끼 - 3)) / 100
	DEBUGPRINTFORML 레즈끼 보정 후：({C_ID}) {캐릭터별판정치}
	;레즈중독에서의 보정 (1+0.1*레즈중독)배
	캐릭터별판정치 = 캐릭터별판정치 * (100 + 10 * ABL:(C_ID):레즈중독) / 100
	DEBUGPRINTFORML 레즈중독 보정 후：({C_ID}) {캐릭터별판정치}

	;궁합에서의 보정
	;사이좋은 캐릭터가 있으면 함께 MASTER를 밀어 넘어뜨리기 쉽고 (완전 취미)
	궁합보정 = 0
	;같은 방의 캐릭터를 살펴 각 캐릭터에 대해 궁합치를 살펴본다
	FOR LOCAL:1, 1, NUM_TARGET + 1
		SIF CFLAG:(TARGET:(LOCAL:1)):동실 != 1
			CONTINUE
		SIF LOCAL == LOCAL:1
			CONTINUE
		;궁합보정 += RELATION:LOCAL:(LOCAL:2)
		;사실은 위의 한 문장만으로 끝내고 싶은데......
		;_Replace.csv의 RELATION 초기값이 작동하지 않아 (울음)
		;그래서 고육지책으로 0은 100으로 간주(폭론)
		궁합보정 += RELATION:(C_ID):(TARGET:(LOCAL:1))
		SIF RELATION:(C_ID):(TARGET:(LOCAL:1)) == 0
			궁합보정 += 100
	NEXT
	;평균치를 취함(소수점 이하 잘라버림) 본인을 제외한 (GET_TARGETNUM() - 1)명으로 나누기
	;0 나누기 주의(없다고 생각하지만)
	궁합보정 /= NUM_TARGET - 1
	캐릭터별판정치 = 캐릭터별판정치 * 궁합보정 / 100
	DEBUGPRINTFORML 궁합보정 후：({(C_ID)}) {캐릭터별판정치}
	;보정 후의 캐릭터별 판정치를 전체 판정치에 넣는다
	판정치 += 캐릭터별판정치
NEXT
;평균치를 취한다(소수점이하로 잘라버림)
;0 나누기 무섭다아
판정치 /= NUM_TARGET
DEBUGPRINTFORML 판정치：{판정치}
;판정값이 최대라 하여, 확률은 「(400+200*동실 인원수)분의 150」으로 설정 (인원수 증가시 난이도 상승)
SIF RAND:(400 + 200 * NUM_TARGET) + 450 <= 판정치
	RETURNF 1
RETURNF 0

;==============================================================
;FileName_ATTITUDE.ERB ------------------------------- Rev1.00
;밀어 넘어뜨려짐 확률 판정
;CALL		USER
;ARG		캐릭터 NO
;RETURN		0:덮치기 불성립, 1:덮치기 성립
;COMMENT	차 흐리기계 함수. 완성도가 나쁘다.

;-----------------------------------------------------------
@ENDURE(ARG)
#FUNCTION
;소질에 따라 판정 변화
LOCAL = CFLAG:ARG:합의판정
LOCAL += (EX:ARG:C절정직전 + EX:ARG:V절정직전 + EX:ARG:A절정직전 + EX:ARG:B절정직전 + EX:ARG:M절정직전) * 20
;껴안으면 밀어 넘어뜨려지는 확률 상승
SIF SELECTCOM == 311
	LOCAL += 30
LOCAL += TALENT:ARG:성적흥미 * 20
SIF TALENT:ARG:처녀
	LOCAL -= 50
SIF TALENT:ARG:연모
	LOCAL = LOCAL + 50
SIF TALENT:ARG:자존심 > 0
	LOCAL -= 20
SIF CFLAG:ARG:생해금 == 2
	LOCAL = LOCAL + 50
SIF CFLAG:ARG:억지로질내사정
	LOCAL -= 100
SIF TALENT:ARG:담력 < 0 || TALENT:ARG:응답 < 0 || TALENT:ARG:무관심
	LOCAL = LOCAL - 30
SIF TALENT:ARG:수치심 > 0
	LOCAL -= 30
SIF TALENT:ARG:수치심 < 0
	LOCAL = LOCAL + 20
SIF TALENT:ARG:자제심
	LOCAL = LOCAL - 20
SIF TALENT:ARG:정조 < 0 || TALENT:ARG:소악마
	LOCAL = LOCAL + 20
SIF TALENT:ARG:정조 > 0 || TALENT:ARG:일선을넘지않음
	LOCAL -= 20
IF !GETBIT(CFLAG:ARG:기정사실, 1) && TALENT:ARG:정조 > 0
	LOCAL -= 30
	SIF !TALENT:ARG:연모 && !TALENT:ARG:섹프
		LOCAL -= 50
ENDIF
SIF OTOKOGIRAI(ARG) && !(IsHeat(ARG) || IsInsenceValid(ARG))
	LOCAL = LOCAL / 2
SIF TALENT:ARG:일선을넘지않음 && TALENT:ARG:연모
	LOCAL = LOCAL * 12 / 5
SIF MARK:ARG:반발각인 == 3
	LOCAL = 0
SIF TFLAG:120 && TFLAG:120 != ARG
	LOCAL = 0
SIF !SHIRAHU(ARG)
	LOCAL = 0
SIF CFLAG:ARG:행동 == 5
	LOCAL = 0
SIF CFLAG:ARG:데일리
	LOCAL = 0
SIF CFLAG:ARG:동침중
	LOCAL = 0
SELECTCASE CFLAG:ARG:욕구불만도
	CASE IS >= 1000
		LOCAL *= 3
	CASE IS >= 800
		LOCAL = LOCAL * 3 / 2
	CASE IS >= 600
		LOCAL = LOCAL * 5 / 4
	CASE IS >= 500
	CASE IS >= 300
		LOCAL /= 2
	CASE IS >= 200
		LOCAL /= 4
	CASE IS >= 100
		LOCAL /= 10
	CASEELSE
		LOCAL = 0
ENDSELECT

;체력에 의존하여 밀어 넘어뜨려짐 확률 감소

SELECTCASE BASE:ARG:체력
	CASE IS < 750
		LOCAL = 0
	CASE IS < 1000
		LOCAL /= 2
	CASE IS < 1500
		LOCAL = LOCAL * 3 / 4
	CASEELSE
ENDSELECT

SELECTCASE BASE:ARG:기력
	CASE 0
		LOCAL = 0
	CASE IS < 500
		LOCAL /= 2
	CASEELSE
ENDSELECT

;이성에 의존하여 밀어 넘어뜨려짐 확률 감소
IF BASE:ARG:이성 > 800
	LOCAL = LOCAL / 5
ELSEIF BASE:ARG:이성 > 600
	LOCAL = LOCAL / 2
ELSE
	;Do Notting
ENDIF
SELECTCASE BASE:ARG:무드
	CASE IS < 200
		LOCAL = 0
	CASE 201 TO 400
		LOCAL /= 4
	CASE 401 TO 600
		LOCAL /= 2
	CASE 601 TO 800
	CASEELSE
		LOCAL = LOCAL * 6 / 5
ENDSELECT
SIF FLAG:생활능력무
	LOCAL = LOCAL * 2
SIF CFLAG:ARG:빡침
	LOCAL = 0
SIF CFLAG:ARG:덮침금지
	LOCAL = 0
SIF TFLAG:펄기한
	LOCAL = 0
RETURNF LOCAL

;==============================================================
;껴안아짐 판정
;==============================================================
@안아다하고판정(ARG)
#FUNCTION
;소질에 따라 판정 변화
LOCAL = CFLAG:ARG:합의판정
SIF TALENT:ARG:연모
	LOCAL = LOCAL + 50
SIF CFLAG:ARG:억지로질내사정
	LOCAL -= 100
;껴안으면 껴안아짐 판정 상승
SIF SELECTCOM == 311
	LOCAL += 30
SIF TALENT:ARG:담력 < 0 || TALENT:ARG:응답 < 0 || TALENT:ARG:무관심 || TALENT:ARG:수치심 > 0
	LOCAL = LOCAL - 30
SIF TALENT:ARG:성적흥미 > 0 || TALENT:ARG:수치심 < 0
	LOCAL = LOCAL + 20
SIF TALENT:ARG:자제심 || TALENT:ARG:성적흥미 < 0
	LOCAL = LOCAL - 20
SIF TALENT:ARG:정조 < 0 || TALENT:ARG:소악마
	LOCAL = LOCAL + 20
SIF TALENT:ARG:정조 > 0 || TALENT:ARG:일선을넘지않음
	LOCAL = LOCAL / 2
SIF OTOKOGIRAI(ARG)
	LOCAL = LOCAL / 2
SIF TALENT:ARG:성별기호 == 3 || MARK:ARG:반발각인 == 3
	LOCAL = 0
SIF TFLAG:120 && TFLAG:120 != ARG
	LOCAL = 0
SIF !SHIRAHU(ARG)
	LOCAL = 0
SIF CFLAG:ARG:행동 == 5
	LOCAL = 0
SIF CFLAG:ARG:빡침
	LOCAL = 0
SIF CFLAG:ARG:데일리
	LOCAL = 0
SIF CFLAG:ARG:동침중
	LOCAL = 0
SELECTCASE CFLAG:ARG:욕구불만도
	CASE IS >= 1000
		LOCAL *= 2
	CASE IS >= 800
		LOCAL = LOCAL * 7 / 5
	CASE IS >= 600
		LOCAL = LOCAL * 6 / 5
	CASE IS >= 400
	CASE IS >= 300
		LOCAL /= 2
	CASE IS >= 150
		LOCAL /= 4
	CASEELSE
		LOCAL = 0
ENDSELECT

;체력에 의존하여 밀어 넘어뜨려짐 확률 감소
SELECTCASE BASE:ARG:체력
	CASE IS < 750
		LOCAL = 0
	CASE IS < 1000
		LOCAL /= 2
	CASE IS < 1500
		LOCAL = LOCAL * 3 / 4
	CASEELSE
ENDSELECT

SELECTCASE BASE:ARG:무드
	CASE IS < 200
		LOCAL = 0
	CASE 201 TO 500
		LOCAL = LOCAL * 2 / 3
	CASE 501 TO 800
	CASEELSE
		LOCAL = LOCAL * 3 / 2
ENDSELECT
SIF FLAG:생활능력무
	LOCAL = LOCAL * 2
RETURNF LOCAL

@HITOKE(동실체크)
;MASTER 위치로 보아 다른 방에 누가 있는지
;동실체크の場合、同室に複数人いる場合も真
#FUNCTION
#DIM 동실체크
SIF 동실체크 && FINDELEMENT(TARGET, 0) - 1 > 1
	RETURNF 1
;공공장소
SIF ROOMDATA:(CFLAG:MASTER:현재위치 % 100) & 장소_공공장소
	RETURNF 1
SIF TFLAG:묘사중인아이
	RETURNF 1
FOR LOCAL,1,CHARANUM
	SIF CFLAG:LOCAL:현재위치 == CFLAG:MASTER:현재위치
		CONTINUE
	SIF CAN_SEE(LOCAL)
		RETURNF 1
NEXT
RETURNF 0

path: ERB/스테이터스계산관련/덮쳐짐_t.ERB


code:
;만취가 되는 처리와 만취로부터 부활하는 처리
@CHARA_ACTION_DRUNK(ARG)
IF !TCVAR:ARG:만취 && BASE:ARG:술기운 > (MAXBASE:ARG:술기운 / 10) * 8
	TCVAR:ARG:만취 = 1

	;일내용이 연회참가의 경우
	IF CFLAG:ARG:직종 == 49
		;일내용을 만취하는으로 변경
		CFLAG:ARG:직종 = 51
		CFLAG:ARG:직장 = TFLAG:연회장
		CFLAG:ARG:행동 = 5
	ELSE
		IF TALENT:ARG:술내성 == -2
			CFLAG:ARG:술을못함 -= 3
		ENDIF
	ENDIF

	IF CFLAG:MASTER:현재위치 == CFLAG:ARG:현재위치
		IF CFLAG:ARG:직종 == 51
			PRINT
		ENDIF
		PRINTFORML %ARG는% 만취했습니다
	ENDIF
ENDIF

;만취간중은 다른 복귀 처리를 사용
IF TCVAR:ARG:만취 && BASE:ARG:술기운 < (MAXBASE:ARG:술기운 / 10) * 4 && !CFLAG:ARG:우후후 && ARG == TARGET
	IF CFLAG:MASTER:현재위치 == CFLAG:ARG:현재위치
		IF CFLAG:ARG:직종 == 51 && CFLAG:ARG:현재위치 == TFLAG:연회장
			PRINT
		ENDIF
		PRINTFORML 만취하고 있던 %CALLNAME:ARG%의 취기가 깬 것 같다
	ENDIF
	CALL 만취부활공통(ARG)
ENDIF

;수간중에 만취로부터 부활할 때의 사전 처리 주인의 소스의 추가
@DRUNK_RECOVER_BEFORE(ARG)
IF TCVAR:ARG:만취 && BASE:ARG:술기운 < (MAXBASE:ARG:술기운 / 10) * 4 && CFLAG:ARG:우후후 && ARG == TARGET && !FLAG:70
	IF TALENT:ARG:연모
		SOURCE:ARG:반감 += 3000
	ELSE
		SOURCE:ARG:반감 += 20000
	ENDIF
	TFLAG:호감도관리 -= MIN(30 - ABL:ARG:친밀, 0) * 3
	TFLAG:신뢰도관리 -= MIN(30 - ABL:ARG:친밀, 0) * 3
	TCVAR:ARG:만취간들킴 = 1
ENDIF

;면간중에 만취로부터 부활하는 처리
@DRUNK_RECOVER(ARG)
SIF !TCVAR:ARG:만취간들킴
	RETURN
PRINTL
PRINTFORMW 만취하고 있던 %ARG가% 눈을 뜬 것 같다!
CALL 만취부활공통(ARG)

CFLAG:ARG:만취간 = 2
CALL ENDUFUFU(ARG)
CALL SET_PRANK(0, ARG, 1)

IF TALENT:ARG:연모
	;MARK:ARG:반발각인 = 1
	;TFLAG:반발각인취득 = 1
	PRINTFORML %ARG는% 흐트러진 의복을 확인하자 날카로운 눈으로 %마스터를% 흘겨 보았다
	PRINTFORMW 복장을 정돈하면서 %ARG는% 차가운 시선을 %CALLNAME:MASTER%에게 보냈다…
ELSE
	;MARK:ARG:반발각인 = 3
	;TFLAG:반발각인취득 = 3
	PRINTFORML %ARG는% 흐트러진 의복을 확인한다고 믿을 수 없다는 표정으로 %마스터를% 응시했다
	PRINTFORMW %ARG는% 분노를 배이게 한 표정을 하고 있다……
ENDIF

@만취부활공통(ARG)
;연회에서 만취하고 있었을 경우
IF CFLAG:ARG:직종 == 51
	CFLAG:ARG:직종 = 0
	TALENT:ARG:기분 = 0
	CFLAG:ARG:연회참가 = 3
ENDIF

TCVAR:ARG:만취 = 0
CFLAG:ARG:행동 = 0

path: ERB/스테이터스계산관련/만취처리_t.ERB


code:
;FileName_SOURCE1.ERB ------------------------------ Rev1. 00
;수면심도의 감소 처리
;ARG		캐릭터 NO
;-----------------------------------------------------------
@수면심도감소(ARG)
SIF CFLAG:MASTER:장난 != 2
	RETURN

LOCAL:1 += CUP:ARG:불쾌 * 10 + CUP:ARG:고통 * 10 + CUP:ARG:공포 + CUP:ARG:반감 + CUP:ARG:억울 / 2
SELECTCASE CUP:ARG:욕정
	CASE 50 TO 100
		LOCAL:1 += 50
	CASE 101 TO 500
		LOCAL:1 += 250
	CASE 501 TO 1000
		LOCAL:1 += 500
	CASE IS > 1000
		LOCAL:1 += 1000
	CASEELSE
ENDSELECT
;		PRINTFORM 수면심도 기초치{LOCAL:1}
SELECTCASE EXP:MASTER:수면간경험
	CASE 100 TO 500
		LOCAL:1 -= 100
	CASE 501 TO 1000
		LOCAL:1 -= 250
	CASE IS > 1000
		LOCAL:1 -= 500
	CASEELSE
ENDSELECT
SELECTCASE EXP:TARGET:수면간경험
	CASE 100 TO 500
		LOCAL:1 -= 100
	CASE 501 TO 1000
		LOCAL:1 -= 250
	CASE IS > 1000
		LOCAL:1 -= 500
	CASEELSE
ENDSELECT
LOCAL:1 = MAX(LOCAL:1, 0)
SIF TCVAR:TARGET:수면제강도 == 1
		LOCAL:1 = LOCAL:1 / 2
SIF TCVAR:TARGET:수면제강도 == 2
		LOCAL:1 = LOCAL:1 / 4
SIF TCVAR:TARGET:수면제강도 == 3
		LOCAL:1 = 0
;PRINTFORM 수면심도가 {LOCAL:1} 줄어들었다
TCVAR:ARG:수면심도 -= LOCAL:1

;FileName_TRACHECK.ERB ----------------------------- Rev1. 00
;호감도 상승 계산 처리
;CALL		USER
;ARG		0:캐릭터 NO
;RETURN		VOID
;COMMENT
;-----------------------------------------------------------
@수면간발각(ARG)
SIF TCVAR:ARG:수면제강도 == 3 && 수면시간(ARG) || FLAG:70
	RETURN
;혼수라고 일어나지 않는다

PRINTFORML %ARG가% 눈을 떴다
IF !TEQUIP:TARGET:밧줄
;밧줄로 묶지 않았다
	IF TALENT:ARG:연모
		PRINTFORML %ARG는% 흐트러진 의복을 확인하자 기가 막힌 것처럼 %마스터를% 응시했다
		PRINTFORMW 복장을 정돈하면서 %ARG는% 적당히 하도록(듯이)라고 잔소리를 했다…
		CFLAG:ARG:호감도 += ABL:ARG:욕망
		PRINTFORML 호감도＋{ABL:ARG:욕망}
		CALL CHARA_AWAKE(ARG, 1,1)
		CALL SET_PRANK(0, ARG, 2)
		CALL ENDUFUFU(ARG)
	ELSEIF TALENT:TARGET:섹프 || TALENT:TARGET:애욕
		PRINTFORML %ARG는% 쾌감에 눈을 뜨자 자신의 몸을 짐승처럼 탐내는 %마스터를% 인식했다
		PRINTFORMW %ARG는% 음미한 포즈를 취하면서도 와 귀여워했으면 좋겠다고 %마스터를% 권하고 있다…
		CFLAG:ARG:호감도 += ABL:ARG:욕망 * 2
		PRINTFORML 호감도＋{ABL:ARG:욕망 * 2}
		CALL CHARA_AWAKE(ARG, 1,1)
		CALL SET_PRANK(0, ARG, 2)
		FOR LOCAL, 1, GET_TARGETNUM() + 1
			CFLAG:(TARGET:LOCAL):우후후 = 1
		NEXT
	ELSE
		PRINTFORML %ARG는% 흐트러진 의복을 확인한다고 믿을 수 없다는 표정으로 %마스터를% 노려보았다
		PRINTFORMW %ARG는% 분노를 배이게 한 표정을 하고 있다……
		IF MARK:ARG:반발각인 <= 2
			MARK:ARG:반발각인 = 3
			PRINTFORML %ARG는% 분노에 얼굴을 비뚤어지게 해, 원망의 말을 중얼거리고 있다……
			SETCOLOR C_RED
			PRINTFORMW %ARG는% 반발각인 LV3를 얻었다
			RESETCOLOR
		ENDIF
		CALL CHARA_AWAKE(ARG, 1,1)
		CALL SET_PRANK(0, ARG, 2)
		CALL ENDUFUFU(ARG)
	ENDIF
ELSE
;밧줄로 묶고 있다면 그대로 약
	IF TALENT:ARG:연모
		PRINTFORML %ARG는% 눈을 뜨자 자신이 묶여져 있어 일을 깨달았다
		PRINTFORMW %마스터를% 눈치채자 기가 막힌 표정으로 빨리 개방 해 주시라고 말했다…
		CALL CHARA_AWAKE(ARG, 1,1)
		CALL SET_PRANK(0, ARG, 2)
		FOR LOCAL, 1, GET_TARGETNUM() + 1
			CFLAG:(TARGET:LOCAL):우후후 = 1
		NEXT
	ELSEIF TALENT:TARGET:섹프 || TALENT:TARGET:애욕
		PRINTFORML %ARG는% 쾌감에 눈을 뜨자 %마스터가% 자신을 묶어 범하고 있는 것에 깨달았다
		PRINTFORMW %ARG는% 허리를 거절하면서도 와 범했으면 좋겠다고 %마스터를% 권하고 있다…
		CFLAG:ARG:호감도 += ABL:ARG:욕망 * 3
		PRINTFORML 호감도＋{ABL:ARG:욕망 * 3}
		CALL CHARA_AWAKE(ARG, 1,1)
		CALL SET_PRANK(0, ARG, 2)
		FOR LOCAL, 1, GET_TARGETNUM() + 1
			CFLAG:(TARGET:LOCAL):우후후 = 1
		NEXT
	ELSE
		PRINTFORML %ARG가% 눈을 뜨자 %CALLNAME:MASTER%에 얽매여 범해지고 있는 것에 깨달았다
		IF TALENT:TARGET:담력 < 0
			PRINTFORMW %ARG는% 울면서 그만두라고 애원하고 있다……
		ELSE
			PRINTFORMW %ARG는% 분노를 배이게 한 표정으로 %마스터를% 매도하고 있다……
		ENDIF
		IF MARK:ARG:반발각인 <= 2
			MARK:ARG:반발각인 = 3
			PRINTFORM %ARG는%
			IF TALENT:ARG:담력 < 0
				PRINTFORM 울면서
			ELSE
				PRINTFORM 분노에 얼굴을 비뚤어지게 해,
			ENDIF
			PRINTFORML 원망의 말을 중얼거리고 있다……
			SETCOLOR C_RED
			PRINTFORMW %ARG는% 반발각인 LV3를 얻었다
			RESETCOLOR
		ENDIF
		CALL CHARA_AWAKE(ARG, 1,1)
		CALL SET_PRANK(0, ARG, 2)
		FOR LOCAL, 1, GET_TARGETNUM() + 1
			CFLAG:(TARGET:LOCAL):우후후 = 1
		NEXT
		TCVAR:ARG:억지로 = 1
	ENDIF
ENDIF
CFLAG:ARG:면간발각 = 1

path: ERB/스테이터스계산관련/수면심도처리_t.ERB


code:
@SHIBORITORI(사정캐릭터번호, 피사정캐릭터번호)
#DIM 사정캐릭터번호
#DIM 피사정캐릭터번호
#DIMS 사정장소1
#DIMS 사정장소2

;어쩐지 DIM라든지 사용해 미묘하게 이해하기 어려울지도 모르지만, 이것은 최초로 사정 확정 후 처리에 넣으려고 역시 멈추었을 때의 아쉬움이므로
;신경쓰지 않아도 됩니다 지문이 최초로 표시되어버린다

IF 사정캐릭터번호 == MASTER
;당신이 돌진하고 있는 경우
	IF S도판정(피사정캐릭터번호, 사정캐릭터번호) < RAND:(MAX((100 - ABL:피사정캐릭터번호:욕망 * 5), 20))
		IF TALENT:피사정캐릭터번호:헌신적 && BASE:사정캐릭터번호:기력 == 0
			PRINTFORMW %조사처리(CALLNAME:피사정캐릭터번호,"는")% %CALLNAME:사정캐릭터번호%의 컨디션을 걱정해, %CALLNAME:사정캐릭터번호%의 페니스를 해방했다……
		ELSE
			PRINTFORMW %조사처리(CALLNAME:피사정캐릭터번호,"는")% %조사처리(CALLNAME:사정캐릭터번호,"가")% 사정이 끝나자, 만족스러운 미소를 지으며 %CALLNAME:사정캐릭터번호%의 페니스를 뽑았다……
		ENDIF
		CALL 삽입해제(피사정캐릭터번호)
		TFLAG:106 = 0
	ELSE
		IF TALENT:피사정캐릭터번호:새드
			IF TALENT:사정캐릭터번호:마조
				IF BASE:피사정캐릭터번호:기력 > 0
					IF BASE:사정캐릭터번호:기력 == 0
						PRINTFORML %조사처리(CALLNAME:피사정캐릭터번호,"는")% %조사처리(CALLNAME:사정캐릭터번호,"를")% 해방할 생각이 전혀 없는 것 같다
						PRINTFORMW %CALLNAME:피사정캐릭터번호%의 질육이 %CALLNAME:사정캐릭터번호%의 페니스에 휘감겨, 한층 더 사정을 요구하며 연동하기 시작하고 있다……
					ELSE
						PRINTFORMW %조사처리(CALLNAME:피사정캐릭터번호,"는")% %CALLNAME:사정캐릭터번호%의 팔을 도망치지 못하도록 단단히 잡고, 내리치듯 허리를 흔들기 시작했다……
					ENDIF
				ELSEIF BASE:사정캐릭터번호:기력 == 0
					PRINTFORMW %조사처리(CALLNAME:피사정캐릭터번호,"는")% 축 늘어진 %조사처리(CALLNAME:사정캐릭터번호,"를")% 누르며, 내리치듯 허리를 흔들기 시작했다……
				ELSE
					PRINTFORMW %조사처리(CALLNAME:피사정캐릭터번호,"는")% %조사처리(CALLNAME:사정캐릭터번호,"가")% 사정하는 것을 확인하자, 간발의 차도 없이 허리를 흔들기 시작했다……
				ENDIF
			ELSEIF BASE:피사정캐릭터번호:기력 > 0
				PRINTFORMW %조사처리(CALLNAME:피사정캐릭터번호,"는")% %CALLNAME:사정캐릭터번호%에게서 더욱 정액을 짜내려고 다시 허리를 흔들기 시작했다……
			ELSEIF BASE:사정캐릭터번호:기력 == 0
				PRINTFORMW %조사처리(CALLNAME:피사정캐릭터번호,"는")% 축 늘어진 %조사처리(CALLNAME:사정캐릭터번호,"를")% 누르며, 다시 허리를 흔들기 시작했다……
			ELSE
				PRINTFORMW %조사처리(CALLNAME:피사정캐릭터번호,"는")% %조사처리(CALLNAME:사정캐릭터번호,"가")% 사정하는 것을 확인하자, 간발의 차도 없이 다시 허리를 흔들기 시작했다……
			ENDIF
		ELSE
			IF TALENT:피사정캐릭터번호:헌신적 && BASE:사정캐릭터번호:기력 == 0
				PRINTFORMW %조사처리(CALLNAME:피사정캐릭터번호,"는")% 아직 만족스럽지 않는지, %CALLNAME:사정캐릭터번호%에게 사과하고 나서 다시 허리를 흔들기 시작했다……
			ELSE
				PRINTFORMW %조사처리(CALLNAME:피사정캐릭터번호,"는")% %조사처리(CALLNAME:사정캐릭터번호,"가")% 사정이 끝나자, %CALLNAME:사정캐릭터번호%의 볼에 키스를 하고 나서 다시 허리를 흔들기 시작했다……
			ENDIF
		ENDIF
	ENDIF
ELSE
;당신에게 돌진하고 있는 경우
	IF TCVAR:TARGET:사정한장소플래그 == 1
		사정장소1 = 질내
		사정장소2 = 질
	ELSE
		사정장소1 = 애널
		사정장소2 = 애널
	ENDIF

	IF S도판정(사정캐릭터번호, 피사정캐릭터번호) - 빼내기정력보정(사정캐릭터번호) < RAND:(MAX((100 - ABL:피사정캐릭터번호:욕망 * 5), 20))
		IF TALENT:사정캐릭터번호:헌신적 && BASE:피사정캐릭터번호:기력 == 0
			PRINTFORMW %조사처리(CALLNAME:사정캐릭터번호,"는")% %CALLNAME:피사정캐릭터번호%의 컨디션을 걱정해, %CALLNAME:피사정캐릭터번호%의 %사정장소2%에서 페니스를 뽑았다……
		ELSE
			PRINTFORMW %조사처리(CALLNAME:사정캐릭터번호,"는")% %CALLNAME:피사정캐릭터번호%의 %사정장소1%에 정액을 쏟아 붓자, 만족한 듯이 미소를 지으며 페니스를 뽑았다……
		ENDIF
		CALL 삽입해제(피사정캐릭터번호)
		CALL TOUCH_SET(0, 0, 사정캐릭터번호, 1)
	ELSE
		IF TALENT:사정캐릭터번호:새드
			IF TALENT:피사정캐릭터번호:마조
				IF BASE:사정캐릭터번호:기력 > 0
					IF BASE:피사정캐릭터번호:기력 == 0
						PRINTFORML %조사처리(CALLNAME:사정캐릭터번호,"는")% %조사처리(CALLNAME:피사정캐릭터번호,"를")% 해방할 생각이 전혀 없는 것 같다
						PRINTFORMW %CALLNAME:사정캐릭터번호%의 페니스는 순식간에 경도와 크기로 되찾아, %CALLNAME:피사정캐릭터번호%의 %조사처리(사정장소2,"를")% 찌르기 시작했다……
					ELSE
						PRINTFORMW %조사처리(CALLNAME:사정캐릭터번호,"는")% %CALLNAME:피사정캐릭터번호%의 팔을 도망치지 못하도록 단단히 잡고, 내리치듯 허리를 찔러 넣기 시작했다……
					ENDIF
				ELSEIF BASE:사정캐릭터번호:기력 == 0
					PRINTFORMW %조사처리(CALLNAME:사정캐릭터번호,"는")% 축 늘어진 %조사처리(CALLNAME:피사정캐릭터번호,"를")% 누르며, 부딪치듯이 허리를 찔러 넣기 시작했다……
				ELSE
					PRINTFORMW %조사처리(CALLNAME:사정캐릭터번호,"는")% 사정이 끝나자, 기대듯이 %조사처리(CALLNAME:피사정캐릭터번호,"를")% 껴안고, 다시 허리를 흔들기 시작했다……
				ENDIF
			ELSEIF BASE:사정캐릭터번호:기력 > 0
				PRINTFORMW %조사처리(CALLNAME:사정캐릭터번호,"는")% %CALLNAME:피사정캐릭터번호%의 %사정장소1%에 정액을 더 부어넣으려고 다시 허리를 흔들기 시작했다……
			ELSEIF BASE:피사정캐릭터번호:기력 == 0
				PRINTFORMW %조사처리(CALLNAME:사정캐릭터번호,"는")% 축 늘어진 %조사처리(CALLNAME:피사정캐릭터번호,"를")% 누르며, 다시 허리를 흔들기 시작했다……
			ELSE
				PRINTFORMW %조사처리(CALLNAME:사정캐릭터번호,"는")% 사정이 끝나자, 기대듯이 %조사처리(CALLNAME:피사정캐릭터번호,"를")% 껴안고, 다시 허리를 흔들기 시작했다……
			ENDIF
		ELSE
			IF TALENT:사정캐릭터번호:헌신적 && BASE:피사정캐릭터번호:기력 == 0
				PRINTFORMW %조사처리(CALLNAME:사정캐릭터번호,"는")% 아직 만족스럽지 않는지, %CALLNAME:피사정캐릭터번호%에게 사과하고 나서 다시 허리를 흔들기 시작했다……
			ELSE
				PRINTFORMW %조사처리(CALLNAME:사정캐릭터번호,"는")% %CALLNAME:피사정캐릭터번호%의 %사정장소1%에 사정하고, %CALLNAME:피사정캐릭터번호%의 볼에 키스를 하고 나서 다시 허리를 흔들기 시작했다……
			ENDIF
		ENDIF
	ENDIF
ENDIF

@S도판정(공격,접수)
#FUNCTION

#DIM 공격
#DIM 접수

VARSET LOCAL

LOCAL += ABL:공격:새드끼 * 5
LOCAL -= ABL:공격:마조끼 * 5
LOCAL += ABL:접수:마조끼 * 5
LOCAL -= ABL:접수:새드끼 * 5

IF TALENT:공격:새드
	IF TALENT:접수:마조
		IF BASE:공격:기력 > 0
			IF BASE:접수:기력 <= 0
				LOCAL += 999
			ELSE
				LOCAL += 90
			ENDIF
		ELSEIF BASE:접수:기력 <= 0
			LOCAL += 75
		ELSE
			LOCAL += 60
		ENDIF
	ELSEIF BASE:공격:기력 > 0
		IF BASE:접수:기력 <= 0
			LOCAL += 50
		ELSE
			LOCAL += 30
		ENDIF
	ELSEIF BASE:접수:기력 <= 0
		LOCAL += 20
	ELSE
		LOCAL += 10
	ENDIF
ELSE
	IF TALENT:접수:마조
		IF BASE:공격:기력 > 0
			IF BASE:접수:기력 <= 0
				LOCAL += 75
			ELSE
				LOCAL += 60
			ENDIF
		ELSEIF BASE:접수:기력 <= 0
			LOCAL += 50
		ELSE
			LOCAL += 35
		ENDIF
	ELSE
		SIF BASE:접수:기력 <= 0
			LOCAL += 10
		SIF BASE:공격:기력 > 0
			LOCAL += 10
	ENDIF
ENDIF

;반항적
SIF TALENT:공격:태도 == 1
	LOCAL += 5

;자제심
SIF TALENT:공격:자제심
	LOCAL -= 10

;쾌감에 솔직
IF TALENT:공격:쾌감응답 == -1
	LOCAL += 10
;쾌감에 솔직
ELSEIF TALENT:공격:쾌감응답 == 1
	LOCAL -= 5
ENDIF

SIF TALENT:공격:헌신적 && BASE:접수:기력 <= 0
	LOCAL /= 5
SIF TALENT:접수:새드
	LOCAL /= 2

RETURNF MAX(LOCAL, 0)

@빼내기정력보정(ARG)
#FUNCTION
RETURNF (MAXBASE:ARG:정력 - BASE:ARG:정력) * 50 / MAXBASE:ARG:정력 + 5

path: ERB/스테이터스계산관련/카운터안의삽입_t.ERB